

# Finiteness in Cubical Type Theory

DONNACHA OISÍN KIDNEY, Imperial College London, United Kingdom

GREGORY PROVAN, University College Cork, Ireland

NICOLAS WU, Imperial College London, United Kingdom

This paper will explore and explain finiteness in constructive mathematics: using this setting, it will also serve as an introduction to constructive mathematics in Cubical Agda, and some related topics.

CCS Concepts: • **Theory of computation** → **Shortest paths; Backtracking; Proof theory; Constructive mathematics; Type theory; Logic and verification**; Algebraic semantics; • **Software and its engineering** → **Functional languages; Data types and structures**.

Additional Key Words and Phrases: Haskell, Agda, graph search, monad

## ACM Reference Format:

Donnacha Oisín Kidney, Gregory Provan, and Nicolas Wu. 2021. Finiteness in Cubical Type Theory. *Proc. ACM Program. Lang.* 5, ICFP, Article 72 (August 2021), 41 pages. <https://doi.org/10.1145/3473577>

## 1 INTRODUCTION

Finiteness in classical mathematics is a simple, almost uninteresting subject. There are many interesting finite things, mind you, and many of those things are finite for interesting reasons; finiteness itself, though, is downright boring.

Part of the reason for the simplicity of finiteness is that there’s really only one notion of finiteness: we say that some set is finite if it has  $n$  elements, where  $n \in \mathbb{N}$ . If we want to *prove* that something is finite we have at our disposal a rich array of techniques and insights: we could show that it has a surjection from another finite set, or that it’s equivalent to another finite set. We could show that any infinite list of its elements must contain duplicates (this is called “streamless”), or that there is a finite list which contains all of its elements. Once proven, though, all of that complexity collapses.

In constructive mathematics the story is a little different. When we prove something constructively, often the thing proven retains components of the proof itself. Take the proposition  $P$  “there is a prime number greater than 100”.

$$P = \exists n. isPrime(n) \wedge n > 100 \quad (1)$$

A classical proof of the above proposition can avoid naming an  $n$ : if we know there are infinite primes, for instance, it is not too hard to deduce that there must be some that are larger than 100.

The constructive existential, on the other hand, is represented by a concrete pair. In the case of the proposition above, the first component of this pair is  $n$ , and the second is a proof that that  $n$  adheres to the predicate  $(isPrime(n) \wedge n > 100)$  in this case. As such, modulo some caveats we

---

Authors’ addresses: [Donnacha Oisín Kidney](mailto:o.kidney21@imperial.ac.uk), Imperial College London, London, United Kingdom, [o.kidney21@imperial.ac.uk](mailto:o.kidney21@imperial.ac.uk); [Gregory Provan](mailto:g.provan@cs.ucc.ie), University College Cork, Cork, Ireland, [g.provan@cs.ucc.ie](mailto:g.provan@cs.ucc.ie); [Nicolas Wu](mailto:n.wu@imperial.ac.uk), Imperial College London, London, United Kingdom, [n.wu@imperial.ac.uk](mailto:n.wu@imperial.ac.uk).

---

Permission to make digital or hard copies of all or part of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. Copyrights for components of this work owned by others than the author(s) must be honored. Abstracting with credit is permitted. To copy otherwise, or republish, to post on servers or to redistribute to lists, requires prior specific permission and/or a fee. Request permissions from [permissions@acm.org](mailto:permissions@acm.org).

© 2021 Copyright held by the owner/author(s). Publication rights licensed to ACM.

2475-1421/2021/8-ART72

<https://doi.org/10.1145/3473577>

will explain soon, there is no way to provide a proof of  $P$  which *doesn't* reveal the particular prime number we had in mind while proving.

This phenomenon is what makes finiteness in constructive mathematics so much richer and more complex than classical finiteness. Proving that something is “streamless” proves something fundamentally different to showing a bijection with a finite set. The other sense in which proofs in constructive mathematics are objects which can be manipulated is that, when these proofs are written on a computer, they often can be *executed* as computer programs. The program “there is a prime number greater than 100” isn’t a terribly useful one, but we will see some examples which are later on.

This thesis will explore and explain finiteness in constructive mathematics: using this setting, it will also serve as an introduction to constructive mathematics in Cubical Agda [Vezzosi et al. 2019], and some related topics.

An exploration of finiteness also provides ample opportunity to explain much of the fundamentals of Homotopy Type Theory [Univalent Foundations Program 2013]: this theory is a foundational system for mathematics, which has at its core the *univalence axiom*. We will dive much more into this axiom later on, but for now know that it gives a formal grounding to say “if two types are isomorphic, it is possible to treat them as the same.” We could, for instance, have two different representations for the natural numbers (binary and unary possibly): the univalence axiom allows us to say “all of the things we have proven about the unary numbers are also true about the binary numbers”. This is an essential technique in most areas of mathematics, which had no formal basis in type theory until recently.

All of our work will be formalised in Agda [Norell 2008], a dependently typed, pure, functional programming language which can be used as a proof assistant. Proof assistants are computer programs which verify the correctness of formal proofs. Proofs in Agda are programs which can be run: its syntax is quite similar to Haskell’s. A recent extension to Agda, Cubical Agda [Vezzosi et al. 2019], allows Agda to compile and typecheck programs written in Cubical Type Theory [Cohen et al. 2016]: this type theory gives a *computational* interpretation of the univalence axiom. By “computational interpretation” here we mean that we will be able to run the programs we write which rely on this axiom.

This thesis is aimed at individuals with some knowledge of Haskell (although extensive knowledge of Haskell is not an absolute necessity) and a curiosity about dependent types. We will explain the basics of dependently-typed programming in Agda, how to write programs and how to write proofs, and we will explain something of the internals of dependent type theory along the way.

## 1.1 Overview

This thesis is structured as follows: in Chapter ?? we will introduce Agda, its syntax, and quickly bring reader up to speed with how to write programs in it. We will also begin to talk about some more foundational type-theory concepts, and we will explain a little about Agda’s particular interpretation of HoTT and CuTT. As we mentioned, some programming experience in Haskell is useful, although not strictly speaking necessary. Certainly no knowledge of dependent type theory is required.

In Chapter 2 we will look in-depth at the focus of this thesis: finiteness. As mentioned finiteness in constructive mathematics is a good deal more complex than finiteness classically: in this thesis alone we will see five separate definitions of “finiteness”, each different, and each with reasonable claim to being the “true” definition of finiteness. We will see which predicates imply each other, what extra information we can derive from the predicates, and what types are included or excluded from each. Along the way we will learn a little more about HoTT and univalence, and we’ll see some practical and direct uses of the univalence axiom.

In Chapter 3 we will look a little at a slightly more advanced application of HoTT: topos theory. A topos is quite a complex, abstract object: it behaves something like the category of sets, although it is more general. For us, showing that sets in Homotopy Type Theory form a topos (well, nearly) will give us access to the general language of toposes.

In Chapter 4 we will combine everything from the previous sections into a (somewhat) practical program for proof search. This program will automatically construct proofs of predicates over finite domains. We will demonstrate the program with the countdown problem [Hutton 2002]: this is a somewhat famous puzzle where players are given five numbers and a target, and have to construct an expression using some or all of the supplied numbers which evaluates to the target.

Finally, in Chapter ?? we will see how we can adapt the work of previous sections to the setting of countably infinite types. Countably infinite are those types that have a bijection with the natural numbers: we will see that we can develop a similar framework of proofs for them as we did for the finite types (with some important differences).

## 1.2 Contributions

Since much of this thesis is dedicated to exploring well-researched topics from a new perspective, some of the contributions can be difficult to tease out. There are three types of valuable and useful work provided in this thesis:

**Exposition of Already-Existing Work** Many explorations of dependent types in programming follow the well-trodden path of describing length-indexed vectors, explaining one or two aspects of totality, and perhaps building up to a proof that  $reverse \circ reverse \equiv id$ . One of the aims of this thesis is to provide a different introduction to dependent types which also touches on some of the core aspects of HoTT, a topic missing much introductory material for functional programmers.

**Formalised Proofs of Theorems proved Informally Elsewhere** “Informal” here means not machine checked. All theorems stated in this thesis are proved formally in Cubical Agda and machine-checked: the full code for these proofs is available online<sup>1</sup>. Formal proofs can be quite different from their informal counterparts, and it is usually not trivial to formalise a pen-and-paper proof. Furthermore, often in formalising a proof new insights about the proof are revealed.

**New Fundamental Theoretical Contributions** Several of the constructions presented in this thesis are in fact novel, and some of the theorems proven have not been proven (either formally or informally) elsewhere.

What follows is an accounting of the specific technical contributions of this thesis:

**A classification of five finiteness predicates in Cubical Type Theory.** Four of these predicates have been previously defined and described in Homotopy Type Theory [Frumin et al. 2018], and two in Martin-Löf type theory [Frumin et al. 2018], this work defines the same predicates in the slightly different setting of Cubical Type Theory. This has the advantage of giving the proofs computational content, a feature used in Section 4. One of the predicates is new (manifest enumerability), and has not been described before.

**Proof of implication between finiteness predicates.** Some of these proofs have been seen in the literature before (from cardinal finiteness to manifest Bishop finiteness is present informally in [Yorgey 2014], and formally (though in HoTT) in [Frumin et al. 2018]). Theorem 2.7 in particular is new, to the best of our knowledge.

**Proof of relation between listed and **Fin**-based finiteness predicates.** There are broadly two different ways to define finiteness predicates: through a list-based approach, or through

<sup>1</sup><https://doisinkidney.com/code/masters-thesis/README.html>

relations with a finite prefix of the natural numbers. Our container-based treatment of the finiteness predicates is novel, to the best of our knowledge, and allows us to prove that the two forms of the predicates are equivalent, which is also novel.

**Proof that Sets form a  $\Pi$ -Pretopos.** This is a proof that has been presented before, in both [Univalent Foundations Program 2013] and [Rijke and Spitters 2015]: here we present the first machine-checked version of this proof.

**Proof that Kuratowski Finite Sets form a  $\Pi$ -Pretopos (Theorem 3.4).** This is a new proof, both formally and informally.

**A library for proof search.** While several libraries for proof search based on finiteness exist in dependently-typed programming languages [Firsov and Uustalu 2015], the one that is presented here is strictly more powerful than those that came before as it is capable of including functions in the search space. This is a direct consequence of the use of Cubical Type Theory: in MLTT functions would not be able to be included at all, and in HoTT the univalence axiom would not have computational content, so a proof search library (structured in the way presented here) would not function. There are other more minor contributions of this library: they are described in detail in subsection 4.

**A verified solver for the countdown problem.** The countdown problem [Hutton 2002] has been studied from a number of angles in functional programming [Bird and Mu 2005]; ours is the first verified solver of the problem.

### 1.3 What is a Proof?

Constructive mathematics is, fundamentally, a different way of thinking about proofs. Classically we have two different universes of things: we have objects, like the natural numbers, or the rings; and the proofs on those things. Constructive mathematics joins those two worlds together.

Practically speaking this means that in constructive mathematics we can't always "prove the existence of" some thing; instead we simply provide that thing. We mentioned an example where this comes into play: to prove that something is finite we might want to prove that it is in bijection with some other finite type. Constructively, this proof *is* a bijection.

Constructive mathematics is also deeply tied to computers. Dependently-typed programming languages, like Agda [Norell 2008], Coq [Team 2020], or Idris [Brady 2013], allow us to compile and *run* our proofs. If we provide a bijection between two types in Agda we also provide a function to and from each type. Agda is therefore both a formal language for constructive proofs and a programming language which can run those proofs.

This relation between proofs and programs is often called the Curry-Howard correspondence. Under this framework, propositions in the logical sense correspond to types in the programming sense [Wadler 2015]; proofs of those propositions then correspond to programs which inhabit those types. As an example we have from propositional logic the following:

$$\frac{A \wedge B}{A}$$

This is conjunction elimination: if we know  $A$  and  $B$ , we also know  $A$ .

Via Curry-Howard, the type corresponding to conjunction is in fact the pair (or tuple).  $A$  and  $B$  are also types. So, the above proposition, as a type, is the following:

$$fst :: (a, b) \rightarrow a$$

And the proof which inhabits this type? It is the function often called `fst`: the selector for the first component of the pair.

$$fst\ (x, y) = x$$

Occasionally we don't want to provide the extra information that constructiveness demands: it is occasionally useful to say “there exists some  $x$  which satisfies some predicate  $P$ ” without revealing the value of  $x$ . In other words, we want to retrieve something of the classical notion of an existential in a constructive setting. Later, for instance, we will begin to work with the category of finite sets. The objects of this category are (of course) finite sets: however, if our notion of “finite” is tied to an explicit bijection, that means that the type of “things which are finite” is pairs of types and bijections between those types. Since there are  $n!$  bijections between a type of cardinality  $n$  and any other, this means that for every such type we have  $n!$  different objects, instead of just 1. We'll see a way to fix this problem with Homotopy Type Theory.

The proofs we will provide in this thesis will be written in the syntax of Agda: they are, however, all valid classical proofs. Constructive mathematics is a subset of classical, after all. When we say “constructive” we really just mean that these proofs avoid reliance on certain axioms like the law of the excluded middle, or the axiom of choice.

## 1.4 Homotopy Type Theory

While this thesis can serve as an introduction to proofs in dependent type theory generally, it can also serve as an introduction to Homotopy Type Theory [Univalent Foundations Program 2013]. HoTT is both a type theory and a foundational theory for mathematics. As a type theory it can be thought of as an alternative to things like Martin-Löf type theory; as a foundational theory it can be compared to other foundational theories like Zermelo–Fraenkel set theory.

*Definition 1 (ZFC).* Zermelo–Fraenkel set theory with choice is a foundational theory for mathematics, which is based on set theory. It is the most commonly-used foundational theory (although it is important to stress that most modern mathematics is foundation-agnostic).

The Axiom of Choice is a non-constructive axiom which is independent of the rest of Zermelo–Fraenkel set theory: when included the theory is often abbreviated to “ZFC”.

A full exploration of ZFC and its axioms is beyond the scope of this thesis; we will, see, however, a definition of the axiom of choice in CuTT (Definition 8).

Central to the theory is the univalence axiom. This axiom states that isomorphism implies equality: more precisely, that “equivalence” is equivalent to equality. We haven't defined equivalence yet (and we haven't defined equality rigorously yet), but the core thrust of the axiom is that it gives proofs of isomorphism all the power of proofs of equality.

From a proof perspective, this is quite useful for transporting proofs from one domain to another: for instance, a proof that some type  $A$  is finite can be transported to a proof that any type isomorphic to  $A$  is finite. From a programming perspective, this is quite useful for transporting *programs*: an API defined on some type can suddenly be reused, without change, on another type as long as the latter is isomorphic to the former. It's worth noting that the technique mentioned here is used quite pervasively in everyday mathematics: it's just that foundational systems (like MLTT) could not justify its use.

The other addition that HoTT gives to traditional type theory is that of Higher Inductive Types. Traditional inductive types are defined by listing their constructors: with HITs we can also list the equalities they must satisfy. The most obvious immediate use of this is quotient types: another concept which is regarded as standard in normal mathematics, but somewhat embarrassingly missing from most constructive theories. In truth, the fact that HITs can be used to describe simple set quotients is almost incidental: in HoTT they are a far more powerful, general tool, which can be used to define a wide variety of essential constructions like the circle, torus, etc. This thesis isn't very interested in these homotopy-focused topics, however it is worth mentioning as it is one of HoTT's great strengths.




Fig. 1. Classification of finiteness predicates according to whether they are discrete (imply decidable equality) and whether they imply a total order.

Finally, we have to explain where Cubical Type Theory [Cohen et al. 2016] fits in. Strictly speaking in this thesis we will not work directly in HoTT: we work instead in CuTT, which is very closely related, but not quite the same, as HoTT. CuTT’s main departure from HoTT is in its representation of equalities, which it calls “paths”. In HoTT (and Martin-Löf type theory, which we will describe later) equalities are an inductive type. In CuTT, paths behave like functions from the real line: an equality  $x \equiv y$  is a function which, when applied to 0, returns  $x$ , and when applied to 1, returns  $y$ .

$$\begin{aligned}
 e : & \quad x \equiv y \\
 e(i) = & \begin{cases} x, & i = 0 \\ \vdots & \\ y, & i = 1 \end{cases} \quad (2)
 \end{aligned}$$

This different “implementation” of paths is central to how CuTT can give computational content to univalence, but it also slightly changes the way paths function.

There is one crucial difference between the two theories: in CuTT the univalence “axiom” is in fact a theorem, with computational content. This means that univalence is not a built-in, assumed to be true, but instead it’s actually derived from other axioms in the system. Practically speaking it means that univalence has computational content: i.e. if we see that two types are equal then we can actually derive the isomorphism that that equality implies. As we’re using CuTT as a programming language (in the form of Cubical Agda [Vezzosi et al. 2019]), this is an essential feature. It means that the proofs we write using univalence still correspond to programs which can be run.



## 2 FINITENESS PREDICATES

In this subsection, we will define and briefly describe each of the five predicates in Figure 1. We will also explain *why* there are five separate predicates: how can it be the case that so many different things describe “finiteness”? As we will see, some predicates are too informative (they tell us more about the underlying type other than it just being finite), or too restrictive (they don’t allow certain finite types to be classified as finite). These diversions won’t be dead-ends, however: the final predicate we will land on as the “correct” (or, more accurately, most useful) notion of finiteness will be built out of all of the others.

### 2.1 Split Enumerability

We will start with a simple notion of finiteness, called split enumerability. This predicate is perhaps the first definition of “finite” that someone might come up with (it’s certainly the most common in dependently-typed programming): put simply, a split enumerable type is a type for which all of its elements can be listed.

*Definition 2 (Split Enumerable Set).* To say that some type  $A$  is split enumerable is to say that there is a list  $\text{support} : \text{List } A$  such that any value  $x : A$  is in  $\text{support}$ .

$$\mathcal{E}! A = \Sigma [\text{support} : \text{List } A] ((x : A) \rightarrow x \in \text{support})$$

We call the first component of this pair the “support” list, and the second component the “cover” proof. An equivalent version of this predicate was called `Listable` in [Firsov and Uustalu \[2015\]](#).

This predicate is simple and useful, but we will see later on how it is perhaps a little imprecise. Before we dive in to exploring the predicate itself, though, we will need to explain some of the terms we used in its definition.

*What is a List?* In this paper we prefer a slightly unusual definition for the type of lists:

$$(3) \quad \text{List} = [\mathbb{N}, \text{Fin}]$$

This is the definition for a *container* (Definition 3): effectively, the above definition says that “Lists are a datatype whose shape is given by the natural numbers, and which can be indexed by numbers smaller than its shape”.

If that seems needlessly complex, don’t worry: this definition is precisely equivalent to the usual inductive one.

```
data List (A : Type a) : Type a where
  [] : List A
  _::_ : A → List A → List A
```

And this isn’t some kind of hand-waving equivalence, either: since we are working in HoTT, we can (and do) prove that the two types are equal, allowing us to use one or the other depending on whichever is more convenient, and `subst` in the other representation without loss of generality. That said, defining lists as containers will reveal several interesting connections and proofs about split enumerability and the other predicates, so for the remainder of the paper whenever we say `List` we will mean Equation 3.

<sup>1</sup>To be absolutely correct we should not say that CuTT implements HoTT, as the two have some subtle differences. While everything we can prove in HoTT can also be proven in CuTT, it is not the case that one is a strict subset of the other. CuTT has many of the same features of HoTT, like HITs and univalence, so almost all of the theory of HoTT applies to both (and indeed almost all of the univalent foundations program applies to both), but there are some slight differences between the theories which mean that some proofs which are valid HoTT are not valid in CuTT.

`Fin zero`  
`Fin (suc`  
 Listing  
 4: Finite  
 Prefixes  
 of  $\mathbb{N}$

We still must define containers themselves, of course. Containers are a well-studied topic in dependent type theory, with a rich theory: we won't dive in to that here.

*Definition 3 (Containers).* A container [Abbott et al. 2005] is a pair  $S, P$  where  $S$  is a type, the elements of which are called the *shapes* of the container, and  $P$  is a type family on  $S$ , where the elements of  $P(s)$  are called the *positions* of a container. We “interpret” a container into a functor defined like so:

$$(5) \quad \llbracket S, P \rrbracket X = \Sigma [s : S] (P s \rightarrow X)$$

The definition of container is a little abstract: it is instructive to think of it more concretely for the case of lists. The container representing finite lists is a pair of a natural number  $n$  representing the length (or “shape”) of the list, and a function  $\text{Fin } n \rightarrow A$ , representing the indexing function into the list.

One of the nice things about containers is it gives us a generic way to define “membership”:

$$(6) \quad x \in xs = \text{fiber } (\text{snd } xs) x$$

Here we're using the homotopy-theory notion of a **fiber** to define membership: a fiber for some function  $f$  and some point  $y$  in its codomain is a value  $x$  and a proof that  $f x \equiv y$ . Membership  $x \in xs$  also makes more sense when described concretely in terms of lists:  $x \in xs$  means “there is an index into  $xs$  such that the index points at an item equal to  $x$ ”.

**Listing 7:** A Fiber [Univalent Foundations Program 2013, definition 4.2.4] *Split Surjections.* Now that we have our terms defined, let's look a little at how split enumerability relates to more traditional, classical notions of finiteness. In a classical setting we likely wouldn't mention “lists” or the like, and would instead define finiteness based on the existence of some injection or surjection, say a surjection from a finite prefix of the natural numbers. In HoTT, surjections (or, more precisely, *split* surjections [Univalent Foundations Program 2013, definition 4.6.1]), are defined like so:

$$\text{SplitSurjective } f = \forall y \rightarrow \text{fiber } f y \quad A \twoheadrightarrow! B = \Sigma (A \rightarrow B) \text{ SplitSurjective}$$

As it turns out, our definition of finiteness here is precisely the same as a surjection-based one, in quite a deep way!

**LEMMA 2.1.** *A proof of split enumerability is equivalent to a split surjection from a finite prefix of the natural numbers.*

$$\mathcal{E}! A \Leftrightarrow \Sigma [n : \mathbb{N}] (\text{Fin } n \twoheadrightarrow! A)$$

**PROOF.**

$\mathcal{E}! A$	$\cong \langle \rangle$	Def. 2 ( $\mathcal{E}!$ )
$\Sigma [xs : \text{List } A] ((x : A) \rightarrow x \in xs)$	$\cong \langle \rangle$	Eqn. 6 ( $\in$ )
$\Sigma [xs : \text{List } A] ((x : A) \rightarrow \text{fiber } (\text{snd } xs) x)$	$\cong \langle \rangle$	Eqn. 8
$\Sigma [xs : \text{List } A] \text{SplitSurjective } (\text{snd } xs)$	$\cong \langle \rangle$	Eqn. 3 (List)
$\Sigma [xs : \llbracket \mathbb{N}, \text{Fin} \rrbracket A] \text{SplitSurjective } (\text{snd } xs)$	$\cong \langle \rangle$	Eqn. 5
$\Sigma [xs : \Sigma [n : \mathbb{N}] (\text{Fin } n \rightarrow A)] \text{SplitSurjective } (\text{snd } xs) \cong \langle \text{reassoc} \rangle$	$\cong \langle \rangle$	Reassociation
$\Sigma [n : \mathbb{N}] \Sigma [f : (\text{Fin } n \rightarrow A)] \text{SplitSurjective } f$	$\cong \langle \rangle$	Eqn. 8
$\Sigma [n : \mathbb{N}] (\text{Fin } n \twoheadrightarrow! A)$	$\blacksquare$	

In the above proof syntax the  $\cong \langle \rangle$  connects lines which are definitionally equal, i.e. they are “obviously” equal from the type checker's perspective. Clearly, only one line isn't a definitional equality:



`reassoc :  $\Sigma (\Sigma A B) C \Leftrightarrow \Sigma [x : A] \Sigma [y : B x] C (x, y)$`

This means that we could have in fact written the whole proof as follows:

`split-enum-is-split-surj :  $\mathcal{E}! A \Leftrightarrow \Sigma [n : \mathbb{N}] (\text{Fin } n \twoheadrightarrow! A)$`   
`split-enum-is-split-surj = reassoc`

The simplicity of this proof, by the way, is why we preferred the container-based definition of lists over the traditional one.

In [Firsov and Uustalu \[2015\]](#), there is a proof that split enumerability and surjections from `Fin` are propositionally equivalent; i.e. a function from each to the other is provided. The fact that the two proofs are precisely equivalent is not proven (and, indeed, it is impossible to prove in MLTT).

*Instances.* To actually show that a type  $A$  is finite amounts to constructing a term of type  $\mathcal{E}! A$ . For simple types like `Bool`, that is simple: it just amounts to basically listing the constructors. As a slightly more complex example, consider the `Fin` type we've been using. Remember that split enumerability is in fact the same as a split surjection from `Fin` (Lemma 2.1): to show that `Fin` is split enumerable, then, we need only show that it has a split surjection from itself. We'll prove the following slightly more general statement:

`$\twoheadrightarrow! \text{-id} : A \twoheadrightarrow! A$`   
 `$\twoheadrightarrow! \text{-id} . \text{fst} = \text{id}$`   
 `$\twoheadrightarrow! \text{-id} . \text{snd } y . \text{fst} = y$`   
 `$\twoheadrightarrow! \text{-id} . \text{snd } y . \text{snd } \_ = y$`

*Decidable Equality.* One thing that characterises all split enumerable types is that they are all *discrete*, i.e. they have decidable equality.

`Discrete A = (x y : A)  $\rightarrow$  Dec (x  $\equiv$  y)`

`data Dec (A : Type a) : Type a where`  
`yes : A  $\rightarrow$  Dec A`  
`no :  $\neg$  A  $\rightarrow$  Dec A`

We will see later that this has implications for the space of types we're dealing with, but for now it simply provides a useful function on split enumerable types.

LEMMA 2.2. *Split enumerability implies decidable equality.*

PROOF. To prove that split enumerability implies decidable equality we'll take a quick detour through injections.

`Injective f =  $\forall x y \rightarrow f x \equiv f y \rightarrow x \equiv y$`       `A  $\hookrightarrow$  B =  $\Sigma [f : (A \rightarrow B)]$  Injective f`

These are useful because we know that any type which injects into a discrete type is itself discrete:

`Discrete-pull-inj : A  $\hookrightarrow$  B  $\rightarrow$  Discrete B  $\rightarrow$  Discrete A`  
`Discrete-pull-inj (f, inj)  $\underline{\underline{=}}$   $\lambda x y \rightarrow$`   
`case (f x  $\underline{\underline{=}}$  f y) of`  
 `$\lambda \{ (\text{no } \neg p) \rightarrow \text{no } (\neg p \circ \text{cong } f)$`   
`; (yes p)  $\rightarrow$  yes (inj x y p) }`

And we can turn a split surjection from  $A$  to  $B$  into an injection from  $B$  to  $A$ :

$\mathcal{E}!(2) : \mathcal{E}!$   
 $\mathcal{E}!(2) . \text{fst}$   
 $\mathcal{E}!(2) . \text{snd}$   
 $\mathcal{E}!(2) . \text{snd}$   
 Listing  
 9: Proof  
 of  
 $\mathcal{E}! \text{Bool}$

```

surj-to-inj : (A →! B) → (B → A)
surj-to-inj (f, surj) .fst x = surj x .fst
surj-to-inj (f, surj) .snd x y f1⟨x⟩≡f1⟨y⟩ =
  x          ≡⟨ surj x .snd ⟩
  f(surj x .fst) ≡⟨ cong f f1⟨x⟩≡f1⟨y⟩ ⟩
  f(surj y .fst) ≡⟨ surj y .snd ⟩
  y ■

```

Yielding a simple proof that any type with a split surjection from a discrete type is itself discrete:

```

Discrete-distrib-surj : (A →! B) → Discrete A → Discrete B
Discrete-distrib-surj = Discrete-pull-inj ∘ surj-to-inj

```

Since split enumerability is really just a split surjection from `Fin`, and since we know that `Fin` is discrete, the overall proof resolves quite simply:

```

ℰ!⇒Discrete : ℰ! A → Discrete A
ℰ!⇒Discrete = flip Discrete-distrib-surj discreteFin
              ∘ snd
              ∘ ℰ!⇔Fin→! .fun

```

■

This lemma is also proven in [Firsov and Uustalu \[2015\]](#), although using a significantly different technique.

## 2.2 Manifest Bishop Finiteness

We mentioned in the introduction that occasionally in constructive mathematics proofs will contain “too much” information. With split enumerability we can see an instance of this. Consider the following proof of the finiteness of `bool`:

```

(10)  ℰ!(2) : ℰ! Bool
      ℰ!(2) .fst = [ false , true , false ]
      ℰ!(2) .snd false = 0 , refl
      ℰ!(2) .snd true  = 1 , refl

```

There is an extra `false` at the end of the support list. There’s nothing terribly wrong with that: it is still a valid proof of finiteness, after all, but it does mean that this proof has some extra information which we didn’t necessarily intend to encode.

There is “slop” in the type of split enumerability: there are more distinct values than there are *usefully* distinct values. To reconcile this, we will disallow duplicates in the support list.

This is where manifest Bishop finiteness comes in: this is a definition of finiteness quite similar to split enumerability in other regards, except that it does not allow for duplicates in the support list.

How exactly to prohibit duplicates is the next question. One approach might be to change the definition of `List`, or introduce a new type `NoDupeList`, and use it in the predicate instead. However, this would mean we lose access to the functions we have defined on lists, and we have to change the definition of `∈` as well.

There is a much simpler and more elegant solution: we insist that every *membership proof* must be unique. This would disallow a definition of `ℰ! Bool` with duplicates, as there are multiple

values which inhabit the type  $\text{false} \in [\text{false}, \text{true}, \text{false}]$ . It also allows us to keep most of the split enumerability definition unchanged, just adding a condition to the returned membership proof in the cover proof.

To specify that a value must exist uniquely in HoTT we can use the concept of a *contraction* [Univalent Foundations Program 2013, definition 3.11.1].

$$(11) \quad \text{isContr } A = \Sigma[ x : A ] \forall y \rightarrow x \equiv y$$

A contraction is a type with the least possible amount of information: it represents the tautologies. All contractions are isomorphic to  $\top$ .

By saying that a proof of membership is a contraction, we are saying that it must be *unique*.

$$(12) \quad x \in! xs = \text{isContr } (x \in xs)$$

Now a proof of  $x \in! xs$  means that  $x$  is not just in  $xs$ , but it appears there *only once*.

With this we can define manifest Bishop finiteness:

*Definition 4 (Manifest Bishop Finiteness).* A type is manifest Bishop finite if there exists a list which contains each value in the type once.

$$\mathcal{B} A = \Sigma[ \text{support} : \text{List } A ] ((x : A) \rightarrow x \in! \text{support})$$

The only difference between manifest Bishop finiteness and split enumerability is the membership term: here we require unique membership ( $\in!$ ), rather than simple membership ( $\in$ ). An equivalent version of this predicate was called `ListableNoDup` in Firsov and Uustalu [2015].

We use the word “manifest” here to distinguish from another common interpretation of Bishop finiteness, which we have called cardinal finiteness in this paper: this version of the proof is “manifest” because we have a concrete, non-truncated list of the elements in the proof.

*The Relationship Between Manifest Bishop Finiteness and Split Enumerability.* While manifest Bishop finiteness might seem stronger than split enumerability, it turns out this is not the case. Both predicates imply the other.

Going from manifest Bishop finiteness is relatively straightforward: to construct a proof of split enumerability from one of manifest Bishop finiteness, it suffices to convert a proof of  $x \in! xs$  to one of  $x \in xs$ , for all  $x$  and  $xs$ . Since  $\in!$  is defined as a contraction of  $\in$ , such a conversion is simply the `fst` function.

Going the other direction takes significantly more work.

LEMMA 2.3. *Any split enumerable set is manifest Bishop finite.*

This lemma is proven in Firsov and Uustalu [2015]. We will only sketch the proof here: the “unique membership” condition in  $\mathcal{B}$  means that we are not permitted duplicates in the support list. The first step in the proof, then, is to filter those duplicates out from the support list of the  $\mathcal{E}!$  proof: we can do this using the decidable equality provided by  $\mathcal{E}!$  (Lemma 2.2). From there, we need to show that the membership proof carries over appropriately.

We have now proved that every manifestly Bishop finite type is split enumerable, and vice versa. While the types are not *equivalent* (there are more split enumerable proofs than there are manifest Bishop finite proofs), they are of equal power.

*From Manifest Bishop Finiteness to Equivalence.* We have seen that split enumerability was in fact a split-surjection in disguise. We will now see that manifest Bishop finiteness is in fact an *equivalence* in disguise. We define equivalences as contractible maps [Univalent Foundations Program 2013, definition 4.4.1]:

$$(13) \quad \text{isEquiv } f = \forall y \rightarrow \text{isContr } (\text{fiber } f \ y) \quad A \simeq B = \Sigma [f : (A \rightarrow B)] \text{ isEquiv } f$$

LEMMA 2.4. *Manifest bishop finiteness is equivalent to an equivalence to a finite prefix of the natural numbers.*

$$\mathcal{B} A \Leftrightarrow \exists [n] (\text{Fin } n \simeq A)$$

PROOF.

$\mathcal{B} A$	$\cong \langle \rangle$	Def. 4 ( $\mathcal{B}$ )
$\Sigma [xs : \text{List } A] ((x : A) \rightarrow x \in! xs)$	$\cong \langle \rangle$	Eqn. 12 ( $\in!$ )
$\Sigma [xs : \text{List } A] ((x : A) \rightarrow \text{isContr } (x \in xs))$	$\cong \langle \rangle$	Eqn. 6 ( $\in$ )
$\Sigma [xs : \text{List } A] ((x : A) \rightarrow \text{isContr } (\text{fiber } (\text{snd } xs) \ x))$	$\cong \langle \rangle$	Eqn. 13
$\Sigma [xs : \text{List } A] \text{ isEquiv } (\text{snd } xs)$	$\cong \langle \rangle$	Eqn. 3 (List)
$\Sigma [xs : [\mathbb{N}, \text{Fin}] A] \text{ isEquiv } (\text{snd } xs)$	$\cong \langle \rangle$	Eqn. 5
$\Sigma [xs : \Sigma [n : \mathbb{N}] (\text{Fin } n \rightarrow A)] \text{ isEquiv } (\text{snd } xs)$	$\cong \langle \text{reassoc} \rangle$	Reassociation
$\Sigma [n : \mathbb{N}] \Sigma [f : (\text{Fin } n \rightarrow A)] \text{ isEquiv } f$	$\cong \langle \rangle$	Eqn. 13
$\exists [n] (\text{Fin } n \simeq A) \blacksquare$		

This proof is almost identical to the proof for Lemma 2.1: it reveals that enumeration-based finiteness predicates are simply another perspective on relation-based ones.

Firsov and Ustalu [2015] provides a proof of the related statement; that manifest Bishop finiteness implies an equivalence to a prefix of the natural number (and vice versa), but similarly to split enumerability they cannot show that the two are equivalent as we have done here.

As we are working in CuTT, a proof of equivalence between two types gives us the ability to *transport* proofs from one type to the other. This is extremely powerful, as we will see.

### 2.3 Cardinal Finiteness

While we have removed some of the unnecessary information from our finiteness predicates, one piece still remains. The two following proofs are both valid proofs of the finiteness of `Bool`, and both do not include any duplicates:

$\mathcal{E}!\langle 2 \rangle : \mathcal{E}! \text{ Bool}$	$\mathcal{E}!\langle 2 \rangle' : \mathcal{E}! \text{ Bool}$
$\mathcal{E}!\langle 2 \rangle .\text{fst} = [\text{false}, \text{true}]$	$\mathcal{E}!\langle 2 \rangle' .\text{fst} = [\text{true}, \text{false}]$
$\mathcal{E}!\langle 2 \rangle .\text{snd } \text{false} = 0, \text{ refl}$	$\mathcal{E}!\langle 2 \rangle' .\text{snd } \text{false} = 1, \text{ refl}$
$\mathcal{E}!\langle 2 \rangle .\text{snd } \text{true} = 1, \text{ refl}$	$\mathcal{E}!\langle 2 \rangle' .\text{snd } \text{true} = 0, \text{ refl}$

Clearly they're not the same though: the order of their support lists differs. Each finiteness predicate so far has contained an *ordering* of the underlying type. For our purposes, this is too much information: it means that when constructing the “category of finite sets” later on, instead of each type having one canonical representative, it will have  $n!$ , where  $n$  is the cardinality of the type<sup>2</sup>.

What we want is a proof of finiteness that is a proposition.

$$(14) \quad \text{isProp } A = (x \ y : A) \rightarrow x \equiv y$$

The mere propositions are one homotopy level higher than the contractions (Equation 11), the types for which all values are equal to some value. They represent the types for which all values

<sup>2</sup>We actually do get a category (a groupoid, even) from manifest Bishop finiteness [Yorgey 2014]: it's the groupoid of finite sets equipped with a linear order, whose morphisms are order-preserving bijections. We do not explore this particular construction in any detail.

are equal, or, the types isomorphic to  $\perp$  or  $\top$ . You can also define propositions in terms of the contractions: propositions are the types whose paths are contractions. Soon (Equation 17) we will see the next homotopy level, which are defined in terms of the propositions.

Despite now knowing the precise property we want our finiteness predicate to have, we're not much closer to achieving it. To remedy the problem, we will use the following type:

$$(15) \quad \begin{aligned} &\text{data } \llbracket \_ \rrbracket (A : \text{Type } a) : \text{Type } a \text{ where} \\ &\quad \_ \llbracket \_ \rrbracket : A \rightarrow \llbracket A \rrbracket \\ &\quad \text{squash} : (x y : \llbracket A \rrbracket) \rightarrow x \equiv y \end{aligned}$$

This is a *higher inductive type*. Normal inductive types have *point* constructors: constructors which construct values of the type. The first constructor here ( $\_ \llbracket \_ \rrbracket$ ), or the constructor `true` for `Bool`, are both “point” constructors.

What makes this type higher inductive is that it also has *path* constructors: constructors which add new equalities to the type. The `squash` constructor here says that all elements of  $\llbracket A \rrbracket$  are equal, regardless of what  $A$  is. In this way it allows us to propositionally truncate types, turning information-containing proofs into mere propositions. Put another way, a proof of type  $\llbracket A \rrbracket$  is a proof that some  $A$  exists, without revealing *which*  $A$ .

To actually use values of this type we have the following eliminator:

$$(16) \quad \text{rec} : \text{isProp } B \rightarrow (A \rightarrow B) \rightarrow \llbracket A \rrbracket \rightarrow B$$

This says that we can eliminate into any proposition: interestingly, this allows us to define a monad instance for  $\llbracket \_ \rrbracket$ , meaning we can use things like `do`-notation.

With this, we can define cardinal finiteness:

*Definition 5 (Cardinal Finiteness).* A type  $A$  is cardinally finite if there exists a propositionally truncated proof that  $A$  is manifest Bishop finite or equivalent to a finite prefix of the natural numbers.

$$\mathcal{C} A = \llbracket \mathcal{B} A \rrbracket$$

This predicate is called Bishop finiteness in [Fruin et al. 2018].

*Deriving Uniquely-Determined Quantities.* At first glance, it might seem that we lose any useful properties we could derive from  $\mathcal{B}$ . Luckily, this is not the case: we will show here how to derive decidable equality (Lemma 2.5) and cardinality (Lemma 2.6) out from under the truncation. Those two lemmas are proven in [Yorgey 2014] (Proposition 2.4.9 and 2.4.10, respectively), in much the same way as we have done here. Our contribution for this subsection is simply the formalisation.

First we'll show that decidable equality carries over from manifest Bishop finiteness. Before we do, note that the fact that we can do this says something interesting about propositional truncation: it has computational, or algorithmic, content. That is in contrast to other ways to “truncate” types:  $\neg\neg P$ , for instance, is a way to provide a “proof” of  $P$  without revealing anything about  $P$  in MLTT. No matter how much we prove that a function from  $P$  doesn't care about which  $P$  it got, though, we can never extract any kind of algorithm or computation from  $\neg\neg P$ .

LEMMA 2.5. *Any cardinal-finite set has decidable equality.*

$$\mathcal{C} A \rightarrow \text{Discrete } A$$

PROOF. We already know that manifest Bishop finiteness implies decidable equality; to apply that proof to cardinal finiteness we'll use the eliminator in Equation 16. Our task, in other words, is to prove the following:

### isProp (Discrete A)

To show that this type is a proposition we must show that any two given members of the type are equal, i.e. we are given two proofs of decidable equality on  $A$  and we must show that they are equal. Remember that `Discrete A` is a function of two arguments returning a `Dec` of whether those two arguments are equal or not. By function extensionality, to prove that that is a proposition we have to prove that `Dec ( $\equiv xy$ )` is a proposition. This proof requires that we show that the payload of each of the constructors (`yes` and `no`) are propositions. `no`'s payload is  $\equiv xy \rightarrow \perp$ , which is a proposition because  $\perp$  is a proposition.

`yes` is a little more interesting: its payload is  $\equiv xy$ . How can we prove that the path between  $x$  and  $y$  is a proposition? It turns out that there is a class of types for which all paths are propositions: the *sets*.

$$\text{isSet } A = (x \ y : A) \rightarrow \text{isProp } (x \equiv y)$$

This is the next homotopy level up from the propositions (Equation 14). More importantly, there is an important theorem relating to sets which *also* relates to decidable equality: Hedberg's theorem [Hedberg 1998]. This tells us that any type with decidable equality is a set.

$$\text{Discrete } A \rightarrow \text{isSet } A$$

And of course we know that  $A$  here has decidable equality: we were just given two proofs of that fact at the beginning of this proof!

This suffices to prove that decidable equality is itself a proposition, and therefore that we can apply Equation 16 and the proof that bishop finiteness implies decidable equality to cardinal finiteness, proving our goal. ■

The next thing we can derive from underneath the truncation in cardinal finiteness is a natural number representing the actual cardinality of the finite type. Of course `N` isn't a proposition, so the eliminator in equation 16 won't work for us here. Instead we will use the following:

$$(18) \quad \text{rec} \rightarrow \text{set} : \text{isSet } B \rightarrow (f : A \rightarrow B) \rightarrow (\forall x \ y \rightarrow f x \equiv f y) \rightarrow \| A \| \rightarrow B$$

This says that we can eliminate into a set as long as the function we use doesn't care about which value it's given: formally,  $f$  in this example has to be "coherently constant" [Kraus 2015].

With that, we can move on to the proof:

LEMMA 2.6. *Given a cardinally finite type, we can derive the type's cardinality, as well as a propositionally truncated proof of equivalence with `Fins` of the same cardinality.*

$$\text{cardinality-is-unique} : \mathcal{C} \ A \rightarrow \exists [n] \| \text{Fin } n \simeq A \|$$

PROOF. The high-level overview of our proof is as follows:

$$\text{cardinality-is-unique} = \text{rec} \rightarrow \text{set } \text{card-isSet } \text{alg } \text{const-} \text{alg} \circ \| \text{map} \| \ \mathcal{B} \Rightarrow \text{Fin} \simeq$$

It is the composition of two operations: first, with `map`, we change the truncated proof of manifest bishop finiteness to a proof of equivalence with `fin`.

Then we use the eliminator from Equation 18 with three parameters. The first simply proves that that the output is a set:

$$\text{card-isSet} : \text{isSet } (\exists [n] \| \text{Fin } n \simeq A \|)$$

The second is the function we apply to the truncated value:

$$\begin{aligned} \text{alg} : \Sigma[ n : \mathbb{N} ] (\text{Fin } n \simeq A) &\rightarrow \Sigma[ n : \mathbb{N} ] \parallel \text{Fin } n \simeq A \parallel \\ \text{alg } (n, f \simeq A) &= n, \mid f \simeq A \mid \end{aligned}$$

And the third is a proof that that function is itself coherently constant:

$$\text{const-alg} : (x \ y : \exists[ n ] (\text{Fin } n \simeq A)) \rightarrow \text{alg } x \equiv \text{alg } y$$

The tricky part of the proof is `const-alg`; here we need to show that `alg` returns the same value no matter its input. That output is a pair, the first component of which is the cardinality, and the second the truncated equivalence proof. The truncated proofs in the output are trivially equal by the truncation, so our obligation now has been reduced to:

$$\frac{(n : \mathbb{N}) \quad (p : \text{Fin } n \simeq A) \quad (m : \mathbb{N}) \quad (q : \text{Fin } m \simeq A)}{\equiv nm}$$

Given univalence we have `Fin n ≡ Fin m`, and the rest of our task is to prove:

$$\frac{\equiv \text{Fin } n \text{Fin } m}{\equiv nm}$$

This is a well-known puzzle in dependently-typed programming, and one that has a surprisingly tricky and complex proof. We do not include it here, since it has already been explored elsewhere, but it is present in our formalisation. ■

*Going from Cardinal Finiteness to Manifest Bishop Finiteness.* We know of course that we can convert any proof of manifest Bishop finiteness to a proof of Cardinal finiteness: it's just the truncation function `|_`. It's the other direction which presents a difficulty:

**THEOREM 2.7.** *Any cardinal finite type with a total order is Bishop finite.*

**PROOF.** The proof for this particular theorem is quite involved in the formalisation, so we only give its sketch here.

Our strategy will be to *sort* the support list of the proof for Bishop finiteness, and then prove that the sorting function is coherently constant, thereby satisfying the eliminator in Equation 18. We need to show, in other words, that sorting two support lists from proofs of manifest Bishop finiteness on the same type with the same order always returns the same result. For simplicity's sake we will use insertion sort:

$$\begin{aligned} \text{insert} : E &\rightarrow \text{List } E \rightarrow \text{List } E \\ \text{insert } x \ [] &= x :: [] \\ \text{insert } x (y :: xs) &\text{ with } x \leq? y \\ \dots \mid \text{inl } x \leq y &= x :: y :: xs \\ \dots \mid \text{inr } y \leq x &= y :: \text{insert } x \ xs \end{aligned} \qquad \begin{aligned} \text{sort} : \text{List } E &\rightarrow \text{List } E \\ \text{sort } [] &= [] \\ \text{sort } (x :: xs) &= \text{insert } x (\text{sort } xs) \end{aligned}$$

And we prove that `sort` produces a list which is sorted, and a permutation of its input.

$$\text{sort-sorts} : \forall xs \rightarrow \text{Sorted } (\text{sort } xs) \qquad \text{sort-perm} : \forall xs \rightarrow \text{sort } xs \rightsquigarrow xs$$

We've introduced two new types here: `Sorted` is a predicate enforcing that the given list is sorted, and `↔` is a permutation relation between two lists. We take the definition of permutations from [Danielsson 2012]: two lists are permutations of each other if their membership proofs are all equivalent.

$$xs \rightsquigarrow ys = \forall x \rightarrow (x \in xs) \Leftrightarrow (x \in ys)$$

This definition fits particularly well for two reasons: first, it is defined on containers generically,



which fits well with our finiteness predicates. Secondly, it is extremely straightforward to show that the support lists of any two proofs of manifest Bishop finiteness must be permutations of each other:

$$(xs\ ys : \mathcal{B}\ A) \rightarrow xs.\text{fst} \rightsquigarrow ys.\text{fst}$$

Almost all of the pieces are in place now: we know that the support lists of all proofs of  $\mathcal{B}\ A$  are permutations of each other, and we know that `sort` returns a sorted permutation of its input. The final piece of the puzzle is the following:

$$\text{sorted-perm-eq} : \forall\ xs\ ys \rightarrow \text{Sorted}\ xs \rightarrow \text{Sorted}\ ys \rightarrow xs \rightsquigarrow ys \rightarrow xs \equiv ys$$

If two sorted lists are both permutations of each other they must be equal. Connecting up all the pieces we get the following:

$$\text{perm-invar} : \forall\ xs\ ys \rightarrow xs \rightsquigarrow ys \rightarrow \text{sort}\ xs \equiv \text{sort}\ ys$$

Because we know that all support lists of  $\mathcal{B}\ A$  are permutations of each other this is enough to prove that `sort` is coherently constant, and therefore can eliminate from within a truncation. The second component of the output pair (the cover proof) follows quite naturally from the definition of permutations. ■

To the best of our knowledge, this is the first explicit proof of this theorem.

*Restrictiveness.* So far our explorations into finiteness predicates have pushed us in the direction of “less informative”: however, as mentioned in the introduction, we can *also* ask how *restrictive* certain predicates are. Since split enumerability and manifest Bishop finiteness imply each other we know that there can be no type which satisfies one but not the other. We also know that manifest Bishop finiteness implies cardinal finiteness, but we do *not* have a function in the other direction:

$$(19) \quad \mathcal{C}\ A \rightarrow \mathcal{B}\ A$$

So the question arises naturally: is there a cardinally finite type which is *not* manifest Bishop finite? It turns out the answer is no! The proof of this fact is relatively short:

$$\begin{aligned} \neg\langle \mathcal{C} \cap \mathcal{B}^c \rangle &: \neg \Sigma[ A : \text{Type } a ]\ \mathcal{C}\ A \times \neg \mathcal{B}\ A \\ \neg\langle \mathcal{C} \cap \mathcal{B}^c \rangle (\_, c, \neg b) &= \text{rec isProp } \perp \neg b\ c \end{aligned}$$

We can apply the function of type  $\mathcal{B}\ A \rightarrow \perp$  (i.e.  $\neg \mathcal{B}\ A$ ) to the value of type  $\|\mathcal{B}\ A\|$  (i.e.  $\mathcal{C}\ A$ ) using Equation 16, since  $\perp$  is itself a proposition. This tells us that manifest bishop finiteness, cardinal finiteness, and split enumerability all refer to the same class of types.

Interestingly, while we cannot construct a function with the type in Equation 19, it does exist *classically*. In fact we can derive it from Equation 20 using the classical monad we developed in the introduction, since Equation 20 is actually equivalent to classical implication.

```

classical-impl :  $\neg (A \times \neg B) \rightarrow \text{Classical } (A \rightarrow B)$ 
classical-impl  $\neg A \times \neg B = \text{do}$ 
  A?  $\leftarrow \text{lem } \{A = A\}$ 
  B?  $\leftarrow \text{lem } \{A = B\}$ 
  case (A?, B?) of
     $\lambda \{ (\text{inl } a, \text{inl } b) \rightarrow \text{pure } (\text{const } b)$ 
      ;  $(\text{inl } a, \text{inr } \neg b) \rightarrow \perp\text{-elim } (\neg A \times \neg B (a, \neg b))$ 
      ;  $(\text{inr } \neg a, \text{inl } b) \rightarrow \text{pure } (\text{const } b)$ 
      ;  $(\text{inr } \neg a, \text{inr } \neg b) \rightarrow \text{pure } (\lambda x \rightarrow \perp\text{-elim } (\neg a x))$ 
    }

```

## 2.4 Manifest Enumerability

Given that we have just proven that all of our finiteness predicates apply to the same types, the natural next step is to try find a predicate which applies to a different class of types. Let's first talk about what this new class of types might look like: what we're looking for is a type which is in some sense finite, but doesn't conform to any of the predicates we've seen so far. The *circle* (Listing 21) is such a type. The thing that this type has which precludes it from being, say, split enumerable, is its *higher homotopy structure*.

So far we have seen three levels of homotopy structure: the contractions (Equation 11), the propositions (Equation 14), and the sets (Equation 17). You may have noticed the pattern that each new level is generated by saying its paths are members of the previous level; if we apply that pattern again, we get to the next homotopy level: the groupoids.

(22)  $\text{isGroupoid } A = (x \ y : A) \rightarrow \text{isSet } (x \equiv y)$

These types do not necessarily have unique identity proofs: there is more than one value which can inhabit the type  $x \equiv y$ . The circle is one of the simplest examples of non-set groupoids: the constructor `loop` is the extra path in the type which isn't the identity path.

We now need to recall two facts: first, Hedberg's theorem tells us that every discrete type is a set. Second, every finiteness predicate we've seen thus far implies decidable equality. From this it's clear that all of the previous predicates are restricted to sets, and can't include types like the circle.

But the type certainly *seems* finite! It has finitely many points, for instance. In order to explore the "restrictiveness" axis in Figure 1, then, we'll need to construct a predicate which admits the circle. Manifest enumerability is one such predicate.

*Definition 6 (Manifest Enumerability).* Manifest enumerability is an enumeration predicate like Bishop finiteness or split enumerability with the only difference being a propositionally truncated membership proof.

$$\mathcal{E} A = \Sigma [ \text{support} : \text{List } A ] ((x : A) \rightarrow \parallel x \in \text{support} \parallel)$$

This predicate is novel, to the best of our knowledge.

It might not be immediately clear why this definition of enumerability allows the circle to conform while the others do not. The crux of the issue was that the cover proofs of the previous definitions didn't just tell us that some element was in the support list, they told us *where* it was in the support list. From the position we were able to derive decidable equality: that position is precisely what's hidden in manifest enumerability.

And indeed this means that the circle is manifestly enumerable.

data S<sup>1</sup> :  
 base : S<sup>1</sup>  
 loop : base == base  
 Listing  
 21: The  
 Circle

$$\begin{aligned}
\mathcal{E}\langle S^1 \rangle &: \mathcal{E} S^1 \\
\mathcal{E}\langle S^1 \rangle .fst &= [ \text{base} ] \\
\mathcal{E}\langle S^1 \rangle .snd &= \llbracket \text{map} \rrbracket (0, \_) \circ \text{isConnectedS}^1
\end{aligned}$$

We use a lemma here, proven in the Cubical Agda library, that  $S^1$  is *connected*:

$$\text{isConnectedS}^1 : (s : S^1) \rightarrow \llbracket \text{base} \equiv s \rrbracket$$

*Surjections.* We already saw that split enumerability was the listed form of a split surjection: what we didn't explain was why the word “split” was placed before surjection. In the presence of higher homotopies than sets, split surjections are actually *not* a satisfactory definition of surjection. And we are most certainly in the presence of higher homotopies: just moments ago we were introduced to the circle. In these cases, the following definition of surjections is preferred [Univalent Foundations Program 2013, definition 4.6.1]:

$$(23) \quad \text{Surjective } f = \forall y \rightarrow \llbracket \text{fiber } f y \rrbracket \qquad A \twoheadrightarrow B = \Sigma (A \rightarrow B) \text{ Surjective}$$

Much in the same way that split enumerability were split surjections, our new predicate of manifest enumerability corresponds to the proper surjections.

LEMMA 2.8. *Manifest enumerability is equivalent to a surjection from a finite prefix of the natural numbers.*

$$\mathcal{E} A \Leftrightarrow \Sigma [ n : \mathbb{N} ] (Fin\ n \twoheadrightarrow A)$$

*Relation To Split Enumerability.* It is trivially easy to construct a proof that any split enumerable type is manifest enumerable: we simply truncate the membership proof. Going the other way is more difficult, as we need to extract the membership proof from under a truncation. We do know what we need, however: the key difference between manifest enumerability and split enumerability is that the latter implied decidable equality. So that's the missing piece we should require in order to go from one to the other:

LEMMA 2.9. *A manifestly enumerable type with decidable equality is split enumerable.*

Now that we know what extra bit of information we are allowed use in this proof, the path forward becomes a little more clear. In terms of the actual conversion function, the support list will stay the same, and only the return type of the cover proof needs to change: from  $\llbracket x \in xs \rrbracket$  to  $x \in xs$ .

That can be accomplished with the help of the following function:

$$\begin{aligned}
(24) \quad \text{recompute} &: \text{Dec } A \rightarrow \llbracket A \rrbracket \rightarrow A \\
\text{recompute } (\text{yes } p) &= p \\
\text{recompute } (\text{no } \neg p) &= \perp\text{-elim } (\text{rec isProp } \perp \neg p)
\end{aligned}$$

Given a decision procedure for some type, and a propositionally truncated value of that type, we can construct an element of the type.

In the case of  $x \in xs$  we can construct a decision procedure for membership of a list, since we already have decidable equality on the elements of the list, proving our obligation.

## 2.5 Kuratowski Finiteness

We now finally arrive at the most important definition of finiteness: Kuratowski finiteness. As a definition, it is quite different from the predicates we've seen (it doesn't involve lists, for instance), but it plays a much larger role in the literature on finiteness predicates than, say, manifest enumerability.

We start with the definition of Kuratowski-finite subsets.

(25) 
$$\begin{aligned} \text{data } \mathcal{K} (A : \text{Type } a) : \text{Type } a \text{ where} \\ \quad [] : \mathcal{K} A \\ \quad \_::\_ : A \rightarrow \mathcal{K} A \rightarrow \mathcal{K} A \\ \text{com} : \forall x y xs \rightarrow x :: y :: xs \equiv y :: x :: xs \\ \text{dup} : \forall x xs \rightarrow x :: x :: xs \equiv x :: xs \\ \text{trunc} : \text{isSet } (\mathcal{K} A) \end{aligned}$$

The first two constructors are point constructors, giving ways to create values of type  $\mathcal{K} A$ . They are also recognisable as the two constructors for finite lists, a type which represents the free monoid. The next two constructors add extra paths to the type: equations that usage of the type must obey. These extra paths turn the free monoid into the free *commutative* (**com**) *idempotent* (**dup**) monoid. The final constructor truncates the type  $\mathcal{K} A$  to a set.

The Kuratowski finite subset is a free join semilattice (or, equivalently, a free commutative idempotent monoid). More prosaically,  $\mathcal{K}$  is the abstract data type for finite sets, as defined in the Boom hierarchy [Boom 1981; Bunkenburg 1994]. However, rather than just being a specification,  $\mathcal{K}$  is fully usable as a data type in its own right, thanks to HITs.

Other definitions of  $\mathcal{K}$  exist (such as the one in [Frumin et al. 2018]) which make the fact that  $\mathcal{K}$  is the free join semilattice more obvious. We have included such a definition in our formalisation, and proven it equivalent to the one above.

$$\begin{aligned} \text{data } \mathcal{K} (A : \text{Type } a) : \text{Type } a \text{ where} \\ \quad \eta : A \rightarrow \mathcal{K} A \\ \quad \_ \cup \_ : \mathcal{K} A \rightarrow \mathcal{K} A \rightarrow \mathcal{K} A \\ \quad \emptyset : \mathcal{K} A \\ \text{U-assoc} : \forall xs ys zs \rightarrow (xs \cup ys) \cup zs \equiv xs \cup (ys \cup zs) \\ \text{U-commutative} : \forall xs ys \rightarrow xs \cup ys \equiv ys \cup xs \\ \text{U-idempotent} : \forall xs \rightarrow xs \cup xs \equiv xs \\ \text{U-identity} : \forall xs \rightarrow xs \cup \emptyset \equiv xs \\ \text{trunc} : \text{isSet } (\mathcal{K} A) \end{aligned}$$

Next, we need a way to say that an entire type is Kuratowski finite. For that, we will need to define membership of  $\mathcal{K}$ .

$$\begin{aligned} x \in [] &= \perp \\ x \in y :: ys &= \parallel xy \uplus x \in ys \parallel \end{aligned}$$

The **com** and **dup** constructors are handled by proving that the truncated form of  $\uplus$  itself commutative and idempotent. The type of propositions is itself a set, satisfying the **trunc** constructor. This gives us enough to define Kuratowski finiteness.

*Definition 7 (Kuratowski Finiteness).* A type is Kuratowski finite if there exists a Kuratowski-finite subset of that type which contains every element of the type.

$$\mathcal{K}^f A = \Sigma [ xs : \mathcal{K} A ] ((x : A) \rightarrow x \in xs)$$

While Kuratowski finiteness is something of the standard formal definition of finiteness, it is quite separated from the enumeration-based definitions we have presented so far. It's difficult to relate to surjections and equivalences, and requires a different style of proof to reason about. As such, we want to get *away* from Kuratowski finiteness as quickly as possible. To do so we use the following lemma:

LEMMA 2.10. *Kuratowski finiteness is equivalent to truncated manifest enumerability.*

$$\| \mathcal{E} A \| \Leftrightarrow \mathcal{K}^f A$$

PROOF. This proof is constructed by providing a pair of functions, to and from each side of the equivalence. This pair implies an equivalence, because both source and target are propositions. This proof, as well as its auxiliary lemmas, are also provided in [Frumin et al. \[2018\]](#), although there the setting is HoTT rather than CuTT. ■

By relating Kuratowski finiteness—with a full equivalence, no less—to an enumerated predicate, we have made it possible to talk about Kuratowski finiteness without interacting with the type at all.

In the next subsection, we will explore the category of discrete Kuratowski finite sets. Under the hood, however, we will really be working with cardinal finite sets. We can do this in a fully rigorous way because Lemma 2.10 allows us to prove the following:

$$(26) \quad \mathcal{C} A \Leftrightarrow \mathcal{K}^f A \times \text{Discrete } A$$

### 3 TOPOS

In this subsection we will examine the categorical interpretation of finite sets. In particular, we will prove that discrete Kuratowski finite types form a  $\Pi$ -pretopos. A lot of the work for this proof has been done already: we have already proven that discrete Kuratowski finiteness is equivalent to cardinal finiteness (Theorem 26), meaning that we can work with the latter definition which is much simpler to prove things about.

There are two reasons we're interested in the categorical and topos-theoretic interpretation of finite sets: first, it's an important theoretical grounding for finite sets, which allows us to understand them in the context of other set-like constructions. Secondly, and more practically, the language of a topos is (or in our case the  $\Pi$ -pretopos) is a common standard framework for doing mathematics generally. This makes it a good basis for an API for building QuickCheck-like generators, for example.

#### 3.1 Categories in HoTT

At first glance, HoTT seems like a perfect setting for category theory: the univalence axiom identifies isomorphisms with equality, a useful tool for category theory missing from MLTT. While this initial impression is broadly true, the construction of categories in HoTT is unfortunately quite complex and involved.

Much of this subsection is simply a summary of parts of [Univalent Foundations Program \[2013, section 9\]](#). The formal proofs we provide are part translation of those proofs in that section, part from [\[Iversen 2018a\]](#) [\[Hu and Carette 2020\]](#), and part our own.

First, we need to think about the type of objects and arrows. We cannot, unfortunately, leave them unrestricted: because of the potential for higher homotopy in HoTT types, we have to restrict the type of arrows to just the sets. This notion: that of a category with all the usual laws such that arrows are a set, is called a *precategory*.

(27) **record** PreCategory  $\ell_1 \ell_2 : \text{Type} (\ell_{\text{suc}} (\ell_1 \ell_{\sqcup} \ell_2))$  **where**  
**field**  
 Ob : Type  $\ell_1$   
 Hom : Ob  $\rightarrow$  Ob  $\rightarrow$  Type  $\ell_2$   
 Id :  $\forall \{X\} \rightarrow \text{Hom } X X$   
 Comp :  $\forall \{X Y Z\} \rightarrow \text{Hom } Y Z \rightarrow \text{Hom } X Y \rightarrow \text{Hom } X Z$   
 assoc-Comp :  $\forall \{W X Y Z\}$   
      $(f : \text{Hom } Y Z)$   
      $(g : \text{Hom } X Y)$   
      $(h : \text{Hom } W X) \rightarrow$   
     Comp  $f(\text{Comp } g h) \equiv \text{Comp } (\text{Comp } f g) h$   
 Comp-Id :  $\forall \{X Y\} (f : \text{Hom } X Y) \rightarrow \text{Comp } f \text{Id} \equiv f$   
 Id-Comp :  $\forall \{X Y\} (f : \text{Hom } X Y) \rightarrow \text{Comp } \text{Id } f \equiv f$   
 Hom-Set :  $\forall \{X Y\} \rightarrow \text{isSet } (\text{Hom } X Y)$

We will use long arrows to refer to morphisms within a category:

$$\_ \longrightarrow \_ = \text{Hom}$$

From here, we can define a notion of isomorphisms.

(28) **Isomorphism** :  $(X \longrightarrow Y) \rightarrow \text{Type } \ell_2$   
**Isomorphism**  $\{X\} \{Y\} f = \Sigma [ g : Y \longrightarrow X ] ((g \cdot f \equiv \text{Id}) \times (f \cdot g \equiv \text{Id}))$   
 $X \cong Y = \Sigma (X \longrightarrow Y) \text{Isomorphism}$

It's a condition on this type which separates the precategories from the categories: if it satisfies a form of univalence, it the precategory is a full category.

(29) **univalent** :  $\{X Y : \text{Ob}\} \rightarrow (X \equiv Y) \simeq (X \cong Y)$

### 3.2 The Category of Sets

Next we'll look at how to construct the category of sets (in the HoTT sense). Much of this work comes directly from [Rijke and Spitters \[2015\]](#) and [Univalent Foundations Program \[2013, section 10\]](#) (the latter of which is in fact an updated and slightly less detailed version of the former). In particular, our treatment (and definition) of categories and topoi comes directly from those works. We have provided in the formalisation a proof that sets in CuTT form a  $\Pi$ -pretopos: this proof (in HoTT) is in fact the main result of [Rijke and Spitters \[2015\]](#); our contribution is simply the formalisation.

The objects are represented by a  $\Sigma$ :

$$\text{Ob} = \Sigma [ t : \text{Type}_0 ] \text{isSet } t$$

This will be quite similar to our objects for finite sets.

Since sets in HoTT don't form a topos, there are quite a few smaller lemmas we need to prove to get as close as we can (a  $\Pi W$ -pretopos): we won't include them here, other than the closure proofs in the following subsection.

### 3.3 Closure

The two most involved proofs for showing that discrete Kuratowski sets form a  $\Pi$ -pretopos are those proofs that show closure under  $\Pi$  and  $\Sigma$ . We will describe them here.

In [Frumin et al. 2018, Theorem 4.21], Kuratowski finite types are proven to be closed under surjections, products, and sums. Here we prove closure under products and sums, but also functions,  $\Sigma$ , and  $\Pi$  (and furthermore our closure proofs are given on all of the finiteness predicates that they apply to).

*Closure of the Ordered Predicates.* First, we will show that split enumerability (and, by extension, manifest enumerability) are closed under  $\Pi$  and  $\Sigma$ . This is the first stepping stone on our way to prove that cardinal finiteness is closed under the same.

Practically speaking, these proofs also open up a wide number of other closure proofs to us. By proving that dependent products and sums are finite, we get the non-dependent cases for free.

LEMMA 3.1. *Split enumerability is closed under  $\Sigma$ .*

$$\_|\Sigma|\_ : \mathcal{E}! A \rightarrow (\forall x \rightarrow \mathcal{E}! (U x)) \rightarrow \mathcal{E}! (\Sigma A U)$$

PROOF. Our task is to construct the two components of the output pair: the support list, and the cover proof. We'll start with the support list: this is constructed by taking the Cartesian product of the input support lists.

```

sup- $\Sigma$  : List A  $\rightarrow$ 
  ((x : A)  $\rightarrow$  List (U x))  $\rightarrow$ 
  List ( $\Sigma$  A U)
sup- $\Sigma$  xs ys = do x  $\leftarrow$  xs
               y  $\leftarrow$  ys x
               [ x, y ]

```

We use `do` notation here because we're working the list monad: this applies the latter function (`ys`) to every element of the list `xs`, and concatenates the results.

To show that this does indeed cover every element of the target type is a little intricate, but not necessarily difficult. ■

Next we'll look at closure under  $\Pi$ . In MLTT, this is of course not provable: since all of the finiteness predicates we have seen so far imply decidable equality, and since we don't have any kind of decidable equality on functions in MLTT, we know that we won't be able to show that any kind of function is finite; even one like `Bool`  $\rightarrow$  `Bool`.

CuTT is not so restricted. Since we have things like function extensionality and transport, we can indeed prove the finiteness of function types. Our proof here makes use directly of the univalence axiom, and makes use furthermore of all the previous closure proofs.

THEOREM 3.2. *Split enumerability is closed under dependent functions ( $\Pi$ -types).*

$$\_|\Pi|\_ : \mathcal{E}! A \rightarrow ((x : A) \rightarrow \mathcal{E}! (U x)) \rightarrow \mathcal{E}! ((x : A) \rightarrow U x)$$

PROOF. Let  $A$  be a split enumerable type, and  $U$  be a type family from  $A$ , which is split enumerable over all points of  $A$ .

As  $A$  is split enumerable, we know that it is also manifestly Bishop finite (Lemma 2.3), and consequently we know  $A \simeq \text{Fin } n$ , for some  $n$  (Lemma 2.4). We can therefore replace all occurrences



of  $A$  with  $\text{Fin } n$ , changing our goal to:

$$\frac{\mathcal{E}! (\text{Fin } n) \quad ((x : \text{Fin } n) \rightarrow \mathcal{E}! (U x))}{\mathcal{E}! ((x : \text{Fin } n) \rightarrow U x)}$$

We then define the type of  $n$ -tuples over some type family.

$$\begin{aligned} \text{Tuple} &: \forall n \rightarrow (\text{Fin } n \rightarrow \text{Type}_0) \rightarrow \text{Type}_0 \\ \text{Tuple zero} & \quad f = \top \\ \text{Tuple (suc } n) & \quad f = f f 0 \times \text{Tuple } n (f \circ fs) \end{aligned}$$

We can show that this type is equivalent to functions (proven in our formalisation):

$$\text{Tuple } n \ U \Leftrightarrow ((i : \text{Fin } n) \rightarrow U i)$$

And therefore we can simplify again our goal to the following:

$$\frac{\mathcal{E}! (\text{Fin } n) \quad ((x : \text{Fin } n) \rightarrow \mathcal{E}! (U x))}{\mathcal{E}! (\text{Tuple } n \ U)}$$

We can prove this goal by showing that  $\text{Tuple } n \ U$  is split enumerable: it is made up of finitely many products of points of  $U$ , which are themselves split enumerable, and  $\top$ , which is also split enumerable. Lemma 3.1 shows us that the product of finitely many split enumerable types is itself split enumerable, proving our goal. ■

*Closure on Cardinal Finiteness.* Since we don't have a function of type  $\mathcal{C} \ A \rightarrow \mathcal{B} \ A$ , closure proofs on  $\mathcal{B}$  do not transfer over to  $\mathcal{C}$  trivially (unlike with  $\mathcal{E}!$  and  $\mathcal{B}$ ). The cases for  $\perp$ ,  $\top$ , and  $\text{Bool}$  are simple to adapt: we can just propositionally truncate their Bishop finiteness proof.

Non-dependent operators like  $\times$ ,  $\curlywedge$ , and  $\rightarrow$  are also relatively straightforward: since  $\| \_ \|$  forms a monad, we can apply  $n$ -ary functions to values inside it, combining them together.

$$\begin{aligned} \_ \times \_ &: \mathcal{B} \ A \rightarrow \\ &\quad \mathcal{B} \ B \rightarrow \\ &\quad \mathcal{B} \ (A \times B) \end{aligned}$$

Into a truncated context:

$$\begin{aligned} \_ \| \times \_ &: \mathcal{C} \ A \rightarrow \\ &\quad \mathcal{C} \ B \rightarrow \\ &\quad \mathcal{C} \ (A \times B) \\ xs \| \times \| \ ys &= \text{do} \\ &\quad x \leftarrow xs \\ &\quad y \leftarrow ys \\ &\quad | \ x \times y \ | \end{aligned}$$

Unfortunately, for the dependent type formers like  $\Sigma$  and  $\Pi$ , the same trick does not work. We have closure proofs like:

$$\frac{\mathcal{B} \ A \quad ((x : A) \rightarrow \mathcal{B} \ (U x))}{\mathcal{B} \ ((x : A) \rightarrow U x)}$$

If we apply the monadic truncation trick we can derive closure proofs like the following:

$$\frac{\| \mathcal{B} \ A \| \quad \| ((x : A) \rightarrow \mathcal{B} \ (U x)) \|}{\| \mathcal{B} \ ((x : A) \rightarrow U x) \|}$$

However our *desired* closure proof is the following:

$$\frac{\llbracket \mathcal{B} A \rrbracket \quad ((x : A) \rightarrow \llbracket \mathcal{B} (U x) \rrbracket)}{\llbracket \mathcal{B} ((x : A) \rightarrow U x) \rrbracket}$$

They don't match!

The solution would be to find a function of the following type:

$$((x : A) \rightarrow \llbracket \mathcal{B} (U x) \rrbracket) \rightarrow \llbracket (x : A) \rightarrow \mathcal{B} (U x) \rrbracket$$

However we might be disheartened at realising that this is a required goal: the above equation is *extremely* similar to the axiom of choice!

*Definition 8 (Axiom of Choice).* In HoTT, the axiom of choice is commonly defined as follows [Univalent Foundations Program 2013, lemma 3.8.2]. For any set  $A$ , and a type family  $U$  which is a set at all the points of  $A$ , the following function exists:

$$((x : A) \rightarrow \llbracket U(x) \rrbracket) \rightarrow \llbracket (x : A) \rightarrow U(x) \rrbracket$$

Luckily the axiom of choice *does* hold for cardinally finite types, allowing us to prove the following:

LEMMA 3.3. *The axiom of choice holds for finite sets.*

$$\mathcal{C} A \rightarrow ((x : A) \rightarrow \llbracket U(x) \rrbracket) \rightarrow \llbracket (x : A) \rightarrow U(x) \rrbracket$$

PROOF. Let  $A$  be a cardinally finite type,  $U$  be a type family on  $A$ , and  $f$  be a dependent function of type  $\Pi(x : A), \llbracket U(x) \rrbracket$ .

First, since our goal is itself propositionally truncated, we have access to values under truncations: put another way, in the context of proving our goal, we can rely on the fact that  $A$  is manifestly Bishop finite. Using the same technique as we did in Lemma 3.2, we can switch from working with dependent functions from  $A$  to  $n$ -tuples, where  $n$  is the cardinality of  $A$ . This changes our goal to the following:

$$\text{Tuple } n (\llbracket \_ \rrbracket \circ U) \rightarrow \llbracket \text{Tuple } n U \rrbracket \quad (30)$$

Since  $\llbracket \_ \rrbracket$  is closed under finite products, this function exists (in fact, using the fact that  $\llbracket \_ \rrbracket$  forms a monad, we can recognise this function as `sequenceA` from the `Traversable` class in Haskell). ■

This lemma is a well-known folklore theorem.

This gets us all of the necessary closure proofs on  $\mathcal{C}$ , and as a result we know the following theorem.

THEOREM 3.4. *Decidable Kuratowski finite sets form a  $\Pi$ -pretopos.*

### 3.4 The Absence of the Subobject Classifier

It's a little unsatisfying that our topos construction has so many caveats: we have to prove a lot of small, uninteresting lemmas just to get to a  $\Pi$ -pretopos, all because we can't prove the one or two larger, simple lemmas which would show that sets form a topos. So what exactly are we missing?

Well, one of the characteristic features of topos theory is that there are a wide variety of equivalent ways to show that something is a topos (a natural consequence of their being a wide variety of things which qualify as toposes). For the direction we have been going, though, the big missing feature is the *subobject classifier*.

A subobject in this context refers to a subset. In set theory, we can often describe a subset of some set  $A$  with the following notation:

$$\{x \mid x \in A; P(x)\}$$

This is the subset of elements in  $A$  which satisfy some predicate  $P$ .

Type theoretically, the way to express the same would be  $\Sigma A P$ : if we wanted to describe the subset of  $\mathbb{N}$  smaller than 10 we would write  $\Sigma[ n : \mathbb{N} ] n < 10$ . In general, however, this type holds too many elements to properly classify the subsets of the larger set: there may be more than one inhabitant of  $P x$  for any given  $x$ . For *propositions*, however, (i.e. where  $P$  is a proposition),  $\Sigma$  represents a perfectly valid encoding of subsets.

The subobject classifier is an object within the topos (which must be a contraction) which classifies monomorphisms (injections). We can actually show that the “subset” notion we just defined does in fact classify monomorphisms in sets in HoTT (in fact directly through univalence), but at this point we run into our one and only size problem in this thesis. The actual object corresponding to the subobject classifier is the following:

$$\begin{aligned} \text{Prop-univ} &: \text{Type}_1 \\ \text{Prop-univ} &= \Sigma[ t : \text{Type}_0 ] \text{isProp } t \end{aligned}$$

The problem here, crucially, is that the universe level of this type is one higher than the universe level of the types it bounds. In other words, this is *not* an object in our  $\Pi$ -pretopos of sets, where the types are all of universe level 0.

Remember that the purpose of universe levels was to prevent Girard’s paradox. However, there is an axiom which removes universe levels to a certain extent which does *not* imply the paradox: propositional resizing.

*Definition 9 (Propositional Resizing).* The axiom of propositional resizing states that the following two types, for any universe level  $u$ , are equivalent:

$$\Sigma[ t : \text{Type } u ] \text{isProp } t \simeq \Sigma[ t : \text{Type } (\ell\text{suc } u) ] \text{isProp } t$$

If propositional resizing holds, then we *can* in fact construct a subobject classifier, for both sets and finite sets.

Mention  
lem?

## 4 SEARCH

A common theme in dependently-typed programming is that proofs of interesting theoretical things often correspond to useful algorithms in some way related to that thing. Finiteness is one such case: if we have a proof that a type  $A$  is finite, we should be able to search through all the elements of that type in a systematic, automated way.

As it happens, this kind of search is a very common method of proof automation in dependently-typed languages like Agda. Proofs of statements like “the following function is associative”

$$\begin{aligned} \_ \wedge \_ &: \text{Bool} \rightarrow \text{Bool} \rightarrow \text{Bool} \\ \text{false} \wedge \text{false} &= \text{false} \\ \text{false} \wedge \text{true} &= \text{false} \\ \text{true} \wedge \text{false} &= \text{false} \\ \text{true} \wedge \text{true} &= \text{true} \end{aligned}$$

can be tedious: the associativity proof in particular would take  $2^3 = 8$  cases. This is unacceptable! There are only finitely many cases to examine, after all, and we’re *already* on a computer: why not automate it? A proof that  $\text{Bool}$  is finite can get us much of the way to a library to do just that.

Similar automation machinery can be leveraged to provide search algorithms for certain “logic programming”-esque problems. Using the machinery we will describe in this subsection, though,

when the program says it finds a solution to some problem that solution will be accompanied by a formal *proof* of its correctness.

In this subsection, we will describe the theoretical underpinning and implementation of a library for proof search over finite domains, based on the finiteness predicates we have introduced already. The library will be able to prove statements like the proof of associativity above, as well as more complex statements. As a running example for a “more complex statement” we will use the countdown problem, which we have been using throughout: we will demonstrate how to construct a prover for the existence of, or absence of, a solution to a given countdown puzzle.

The API for writing searches over finite domains comes from the language of the  $\Pi$ -pretopos: with it we will show how to compose QuickCheck-like generators for proof search, with the addition of some automation machinery that allows us to prove things like the associativity in a couple of lines:

$$(31) \quad \begin{aligned} \wedge\text{-assoc} &: \forall x y z \rightarrow (x \wedge y) \wedge z \equiv x \wedge (y \wedge z) \\ \wedge\text{-assoc} &= \forall \lambda^n \lambda x y z \rightarrow (x \wedge y) \wedge z \stackrel{?}{=} x \wedge (y \wedge z) \end{aligned}$$

We have already, in previous sections, explored the theoretical implications of Cubical Type Theory on our formalisation. With this library for proof search, however, we will see two distinct practical applications which would simply not be possible without computational univalence. First and foremost: our proofs of finiteness, constructed with the API we will describe, have all the power of full equalities. Put another way any proof over a finite type  $A$  can be lifted to any other type with the same cardinality. Secondly our proof search can range over functions: we could, for instance, have asked the prover to find if *any* function over **Bool** is associative, and if so return it to us.

$$\begin{aligned} \text{some-assoc} &: \Sigma [f : (\text{Bool} \rightarrow \text{Bool} \rightarrow \text{Bool})] \forall x y z \rightarrow f(f x y) z \equiv f x (f y z) \\ \text{some-assoc} &= \exists \lambda^n \lambda f \rightarrow \forall \lambda^n \lambda x y z \rightarrow f(f x y) z \stackrel{?}{=} f x (f y z) \end{aligned}$$

The usefulness of which is dubious, but we will see a more interesting application soon.

#### 4.1 How to make the Typechecker do Automation

For this prover we will not resort to reflection or similar techniques: instead, we will trick the type checker to do our automation for us. This is a relatively common technique, although not so much outside of Agda, so we will briefly explain it here.

To understand the technique we should first notice that some proof automation *already* happens in Agda, like the following:

$$\begin{aligned} \text{obvious} &: \text{true} \wedge \text{false} \equiv \text{false} \\ \text{obvious} &= \text{refl} \end{aligned}$$

The type checker does not require us to manually explain each step of evaluation of **true**  $\wedge$  **false**. While it’s not a particularly impressive example of automation, it does nonetheless demonstrate a principle we will exploit: closed terms will compute to a normal form if they’re needed to type check. The type checker will perform  $\beta$ -reduction as much as it can.

So our task is to rewrite proof obligations like the one in Equation 31 into ones which can reduce completely. As it turns out, we have already described the type of proofs which can “reduce completely”: *decidable* proofs. If we have a decision procedure over some proposition  $P$  we can run that decision during type checking, because the decision procedure itself is a proof that the decision will terminate. In code, we capture this idea with the following pair of functions:

$\text{True} : \text{Dec } A \rightarrow \text{Type}_0$ $\text{True } (\text{yes } \_) = \top$ $\text{True } (\text{no } \_) = \perp$	$\text{toWitness} : (\text{decision} : \text{Dec } A) \rightarrow$ $\quad \{ \_ : \text{True } \text{decision} \} \rightarrow A$ $\text{toWitness } (\text{yes } x) = x$
---	--

The first is a function which derives a type from whether a decision is successful or not. This function is important because if we use the output of this type at any point we will effectively force the unifier to run the decision computation. The second takes—as an implicit argument—an inhabitant of the type generated from the first, and uses it to prove that the decision can only be true, and the extracts the resulting proof from that decision. All in all, we can use it like this:

$\text{extremely-obvious} : \text{true} \neq \text{false}$   
 $\text{extremely-obvious} = \text{from-true } (! (\text{true} \stackrel{?}{=} \text{false}))$

This technique will allow us to automatically compute any decidable predicate.

## 4.2 Omniscience

So we now know what is needed of us for proof automation: we need to take our proofs and make them decidable. In particular, we need to be able to “lift” decidability back over a function arrow. For instance, given  $x$ ,  $y$ , and  $z$  we already have  $\text{Dec } ((x \wedge y) \wedge z \equiv x \wedge (y \wedge z))$  (because equality over booleans is decidable). In order to turn this into a proof that  $\wedge$  is associative we need  $\text{Dec } (\forall x \ y \ z \rightarrow (x \wedge y) \wedge z \equiv x \wedge (y \wedge z))$ . The ability to do this is described formally by the notion of “Exhaustibility”.

$\text{Exhaustible } p \ A = \forall \{P : A \rightarrow \text{Type } p\} \rightarrow (\forall x \rightarrow \text{Dec } (P x)) \rightarrow \text{Dec } (\forall x \rightarrow P x)$

We say a type  $A$  is exhaustible if, for any decidable predicate  $P$  on  $A$ , the universal quantification of the predicate is decidable.

This property of **Bool** would allow us to automate the proof of associativity, but it is in fact not strong enough to find individual representatives of a type which support some property. For that we need the more well-known related property of *omniscience*.

$\text{Omniscient } p \ A = \forall \{P : A \rightarrow \text{Type } p\} \rightarrow (\forall x \rightarrow \text{Dec } (P x)) \rightarrow \text{Dec } (\exists [x] P x)$

The “limited principle of omniscience” [Bishop 1967] is a classical principle which says that omniscience holds for all sets. It doesn’t hold constructively, of course: it lies a little bit below LEM in terms of its non-constructiveness, given that it can be derived from LEM but LEM cannot be derived from it.

Omniscience implies exhaustibility: we can use the usual rule of  $\neg \exists x. P(x) \iff \forall x. \neg P(x)$  to turn omniscience for some predicate  $P$  into exhaustibility for some predicate  $\neg P$ . Usually we don’t have double negation elimination constructively, but since  $P$  is decidable it’s actually present in this case:

$\text{Dec} \rightarrow \text{DoubleNegElim} : (A : \text{Type } a) \rightarrow \text{Dec } A \rightarrow \neg \neg A \rightarrow A$   
 $\text{Dec} \rightarrow \text{DoubleNegElim } A \ (\text{yes } p) \ \_ = p$   
 $\text{Dec} \rightarrow \text{DoubleNegElim } A \ (\text{no } \neg p) \ \text{contra} = \perp\text{-elim } (\text{contra } \neg p)$

All together, this gives us the following proof:

```

Omniscient→Exhaustible : Omniscient p A → Exhaustible p A
Omniscient→Exhaustible omn P? =
  map-dec
    (λ ¬∃P x → Dec→DoubleNegElim _ (P? x) (¬∃P o (x ,_)))
    (λ ¬∃P ∀P → ¬∃P λ p → p .snd (∀P (p .fst)))
    (! (omn (! o P?)))

```

Our focus here is on those types for which omniscience *does* hold, which includes the (ordered) finite types. Perhaps surprisingly, it is not *only* finite types which are exhaustible. Certain infinite types can be exhaustible [Escardo 2007], but an exploration of that is beyond the scope of this work.

All of the finiteness predicates imply exhaustibility. To prove that fact we'll just show that the Kuratowski finite types are exhaustible: since it's the weakest predicate, and can be derived from all the others.

LEMMA 4.1. *Kuratowski finiteness implies exhaustibility.*

Manifest enumerability is similarly the weakest of the ordered predicates:

LEMMA 4.2. *Manifest enumerability implies omniscience.*

We won't provide these full proofs here, since they are rather tedious and don't provide much insight.

Finally, there is a form of omniscience which works with Kuratowski finiteness:

```

Prop-Omniscient p A = ∀ {P : A → Type p} → (∀ x → Dec (P x)) → Dec || ∃[ x ] P x ||

```

By truncating the returned  $\Sigma$  we don't reveal which  $A$  we've chosen which satisfies the predicate: this means that it can be pulled out of the Kuratowski finite subset without issue.

```

Kf⇒Prop-Omniscient : Kf A → Prop-Omniscient p A
Kf⇒Prop-Omniscient K P? =
  PropTrunc.rec
    (isPropDec squash)
    (map-dec | _ | refute-trunc o λ xs → ℒ⇒Omniscient xs P?)
    (Kf⇒||ℒ|| K)

```

With the knowledge that any Kuratowski finite type implies exhaustibility we know that we can do proof search over all of the types we have proven to be Kuratowski finite: the 0, 1, and 2 types; (dependent) sums and products; and any type proven to be equivalent to these. It's still not entirely clear how to actually *use* this automation without incurring so much boilerplate as to defeat the point, though.

### 4.3 An Interface for Proof Automation

In this subsection we will present the more user-friendly interface to the library, designed to be used to automate away tedious proofs in an easy way.

*The Design of the Interface.* The central idea of the interface to the proof search library are the following two functions:

$$\begin{array}{ll}
\forall? : \mathcal{E}! A \rightarrow & \exists? : \mathcal{E}! A \rightarrow \\
(\forall x \rightarrow \text{Dec } (P x)) \rightarrow & (\forall x \rightarrow \text{Dec } (P x)) \rightarrow \\
\text{Dec } (\forall x \rightarrow P x) & \text{Dec } (\exists [x] P x) \\
\forall? \mathcal{E}!\langle A \rangle = \mathcal{E}! \Rightarrow \text{Exhaustible } \mathcal{E}!\langle A \rangle & \exists? \mathcal{E}!\langle A \rangle = \mathcal{E}! \Rightarrow \text{Omniscient } \mathcal{E}!\langle A \rangle
\end{array}$$

Clearly they're just restatements of exhaustibility and omniscience. However, we can combine these functions with the automation technique from above to create the following:

$$\begin{array}{ll}
\forall\downarrow : (\mathcal{E}!\langle A \rangle : \mathcal{E}! A) \rightarrow & \exists\downarrow : (\mathcal{E}!\langle A \rangle : \mathcal{E}! A) \rightarrow \\
(P? : \forall x \rightarrow \text{Dec } (P x)) \rightarrow & (P? : \forall x \rightarrow \text{Dec } (P x)) \rightarrow \\
\llbracket \_ : \text{True } (\forall? \mathcal{E}!\langle A \rangle P?) \rrbracket \rightarrow & \llbracket \_ : \text{True } (\exists? \mathcal{E}!\langle A \rangle P?) \rrbracket \rightarrow \\
\forall x \rightarrow P x & \exists [x] P x \\
\forall\downarrow \_ \llbracket t \rrbracket = \text{toWitness } t & \exists\downarrow \_ \llbracket t \rrbracket = \text{toWitness } t
\end{array}$$

This automation procedure allows us to state the property succinctly, and have the type checker go and run the decision procedure to solve it for us. Here's an example of its use:

$$\begin{array}{l}
\wedge\text{-idem} : \forall x \rightarrow x \wedge x \equiv x \\
\wedge\text{-idem} = \forall\downarrow \mathcal{E}!\langle 2 \rangle \lambda x \rightarrow x \wedge x \stackrel{?}{=} x
\end{array}$$

*Instances.* One bit of cruft in the above proof is the need to specify the particular finiteness proof for bools. While this isn't any great burden in this case, it of course becomes more difficult in more complex circumstances.

To solve this we can use Agda's instance search. This changes the definitions of our automation functions to the following:

$$\begin{array}{ll}
\forall\downarrow : \llbracket \mathcal{E}!\langle A \rangle : \mathcal{E}! A \rrbracket \rightarrow & \exists\downarrow : \llbracket \mathcal{E}!\langle A \rangle : \mathcal{E}! A \rrbracket \rightarrow \\
(P? : \forall x \rightarrow \text{Dec } (P x)) \rightarrow & (P? : \forall x \rightarrow \text{Dec } (P x)) \rightarrow \\
\llbracket \_ : \text{True } (\forall? \mathcal{E}!\langle A \rangle P?) \rrbracket \rightarrow & \llbracket \_ : \text{True } (\exists? \mathcal{E}!\langle A \rangle P?) \rrbracket \rightarrow \\
\forall x \rightarrow P x & \exists [x] P x \\
\forall\downarrow \_ \llbracket t \rrbracket = \text{toWitness } t & \exists\downarrow \_ \llbracket t \rrbracket = \text{toWitness } t
\end{array}$$

And this also changes the idempotency proof to the following:

$$\begin{array}{l}
\wedge\text{-idem} : \forall x \rightarrow x \wedge x \equiv x \\
\wedge\text{-idem} = \forall\downarrow \lambda x \rightarrow x \wedge x \stackrel{?}{=} x
\end{array}
\tag{32}$$

Again, there's not any great revelation in ease of use here, but more complex examples really benefit. Especially when we build the full set of instances: any expression built out of products and sums will automatically have an instance. This will allow us, for instance, to perform proof search over tuples, which gives us some degree of automation for proof search in tuples.

$$\begin{array}{l}
\wedge\text{-comm} : \forall x y \rightarrow x \wedge y \equiv y \wedge x \\
\wedge\text{-comm} = \text{curry } (\forall\downarrow (\text{uncurry } (\lambda x y \rightarrow x \wedge y \stackrel{?}{=} y \wedge x)))
\end{array}$$

These instances aren't limited to non-dependent sums and products, either: for  $\Sigma$ , for instance, we already have a proof that  $\mathcal{E}! A \rightarrow (\forall x \rightarrow \mathcal{E}! (B x)) \rightarrow \mathcal{E}! (\Sigma A B)$ . Since  $A$  is finite, we can construct a finite constraint that “ $B$  is finite at all points of  $A$ ”, and use that to statically build our instance.



```

_ : ℰ! (Σ[ s : Bool ] (if s then Fin 3 else Fin 4))
_ = it

```

The `it` function here is a clever helper function. It's defined like so:

```

it : { _ : A } → A
it { x } = x

```

Basically it searches for an instance for the type in the hole that it's put into: it's a way of asking Agda to “find an instance which fits here”.

*Generic Currying and Uncurrying.* While we have arguably removed the bulk of the boilerplate from the automated proofs, there is still the case of the ugly noise of currying and uncurrying. In this subsection, we take inspiration from Allais [2019] to develop a small interface to generic  $n$ -ary functions and properties. We will describe it briefly here.

The basic idea of currying and uncurrying generically is to allow ourselves to work with a generic and flexible representation of function arguments which can be manipulated more easily than a simple function itself. Our first task, then, is to define that representation of function arguments. As in Allais [2019], our representation is a tuple which is in some sense a “second order” indexed type. By second order here we mean that it is an indexed type indexed by another indexed type. The reason for this complexity is that our solution is to be fully level-polymorphic. To start, we define a type representing a vector of universe levels:

```

Levels : ℕ → Type₀
Levels zero = ⊤
Levels (suc n) = Level × Levels n

max-level : Levels n → Level
max-level {zero} _ = ℓzero
max-level {suc n} (x , xs) =
  x ℓ⊔ max-level xs

```

This will be used to assign our tuple the correct universe level generically. Next, we define the list of types (this type is indexed by the list of universe levels of each type):

```

Types : ∀ n → (ls : Levels n) → Type (ℓsuc (max-level ls))
Types zero ls = ⊤
Types (suc n) (l , ls) = Type l × Types n ls

```

And finally, the tuple, indexed by its list of types:

```

(⟦_⟧)+ : Types (suc n) ls →
  Type (max-level ls)
(⟦_⟧)+ {n = zero} (X , Xs) = X
(⟦_⟧)+ {n = suc n} (X , Xs) = X × (⟦Xs⟧)+

(⟦_⟧) : Types n ls →
  Type (max-level ls)
(⟦_⟧) {n = zero} _ = ⊤
(⟦_⟧) {n = suc n} = (⟦_⟧)+ {n = n}

```

The reason for two separate functions here is to avoid the  $\top$ -terminated tuples we would need if we just had one. This means that, for instance, to represent a tuple of a `Bool` and `ℕ` we can write `(true , 2)` instead of `(true , 2 , tt)`.

Next we turn to how we will represent functions. In Agda there are three ways to pass function arguments: explicitly, implicitly, and as an instance. We will represent these three different versions with a data type:

```

data ArgForm : Type₀ where expl impl inst : ArgForm

```

And then we can make a type for functions in the general sense: a type which has this sum type as a parameter.

$$\begin{aligned} \_[\_] \rightarrow \_ &: \text{Type } a \rightarrow \text{ArgForm} \rightarrow \text{Type } b \rightarrow \text{Type } (a \ell \sqcup b) \\ A [\text{expl}] \rightarrow B &= A \rightarrow B \\ A [\text{impl}] \rightarrow B &= \{ \_ : A \} \rightarrow B \\ A [\text{inst}] \rightarrow B &= \llbracket \_ : A \rrbracket \rightarrow B \end{aligned}$$

And we can show that this is isomorphic to a normal function:

$$[\_] \$ : \forall \text{form} \rightarrow (A [\text{form}] \rightarrow B) \Leftrightarrow (A \rightarrow B)$$

This of course is only a representation of *non*-dependent functions. Dependent functions are defined in a similar way:

$$\Pi[\_] \$ : \forall \{B : A \rightarrow \text{Type } b\} \text{fr} \rightarrow (x : A \Pi[\text{fr}] \rightarrow B x) \Leftrightarrow ((x : A) \rightarrow B x)$$

Using both of these things, we can now define a generic type for multi-argument functions:

$$\begin{aligned} (\_) (\_) \rightarrow \_ &: \text{Types } n \text{ ls} \rightarrow \text{ArgForm} \rightarrow \text{Type } \ell \rightarrow \text{Type } (\text{max-level } ls \ell \sqcup \ell) \\ (\_) (\_) \rightarrow \_ \{n = \text{zero}\} \text{Xs fr } Y &= Y \\ (\_) (\_) \rightarrow \_ \{n = \text{suc } n\} (X, \text{Xs}) \text{fr } Y &= X [\text{fr}] \rightarrow \llbracket \text{Xs} \rrbracket [\text{fr}] \rightarrow Y \end{aligned}$$

We can also define multi-argument dependent functions in a similar way. Similarly to how we had to define two tuple types in order to avoid the  $\top$ -terminated tuples, we have two definitions for multi-argument dependent functions. We only include the nonempty version here for brevity.

$$\begin{aligned} \text{pi-arrs-plus} &: \\ & (Xs : \text{Types } (\text{suc } n) \text{ ls}) \rightarrow \\ & \text{ArgForm} \rightarrow \\ & (y : \llbracket Xs \rrbracket^+ \rightarrow \text{Type } \ell) \rightarrow \\ & \text{Type } (\text{max-level } ls \ell \sqcup \ell) \\ \text{pi-arrs-plus } \{n = \text{zero}\} (X, \text{Xs}) \text{fr } Y &= x : X \Pi[\text{fr}] \rightarrow Y x \\ \text{pi-arrs-plus } \{n = \text{suc } n\} (X, \text{Xs}) \text{fr } Y &= \\ & x : X \Pi[\text{fr}] \rightarrow xs : \llbracket Xs \rrbracket^+ \Pi[\text{fr}] \rightarrow Y (x, xs) \end{aligned}$$

Finally, this all allows us to define an isomorphism between generic multi-argument dependent functions and their uncurried forms.

$$\begin{aligned} \Pi[\_] \wedge \_ \$ &: \forall n \{ls \ell\} \text{fr} \{Xs : \text{Types } n \text{ ls}\} \{Y : \llbracket Xs \rrbracket \rightarrow \text{Type } \ell\} \rightarrow \\ & (xs : \llbracket Xs \rrbracket \Pi[\text{fr}] \rightarrow Y xs) \Leftrightarrow ((xs : \llbracket Xs \rrbracket) \rightarrow Y xs) \end{aligned}$$

The use of all of this is that we can take the user-supplied curried version of a function and transform it into a version which takes instance arguments for each of the types.

$$\begin{aligned} \exists?^n &: (\text{map-types } \mathcal{E}! \text{Xs}) [\text{inst}] \rightarrow \\ & xs : \llbracket Xs \rrbracket \Pi[\text{expl}] \rightarrow \\ & \text{Dec } (P \text{xs}) [\text{expl}] \rightarrow \\ & \text{Dec } (\Sigma (\llbracket Xs \rrbracket) P) \\ \exists?^n &= [n \wedge \text{inst } \$] .\text{inv } \lambda fs \\ & \rightarrow \mathcal{E}! \Rightarrow \text{Omniscient } (\text{tup-inst } n fs) \\ & \circ \Pi[n \wedge \text{expl } \$] .\text{fun} \end{aligned}$$

```

 $\exists_{\mathcal{E}}^n :$ 
   $insts : (\text{map-types } \mathcal{E}! Xs) \Pi[inst] \rightarrow$ 
   $((P? : xs : (\text{map-types } \mathcal{E}! Xs) \Pi[expl] \rightarrow \text{Dec } (P xs)))$ 
   $\rightarrow \{ \_ : \text{True} \}$ 
   $(\mathcal{E}! \Rightarrow \text{Omniscient}$ 
     $(\text{tup-inst } n insts)$ 
     $(\Pi[ n \wedge expl \$ ] .fun P?)) \}$ 
   $\rightarrow \Sigma (\text{map-types } \mathcal{E}! Xs) P$ 
 $\exists_{\mathcal{E}}^n =$ 
   $\Pi[ n \wedge inst \$ ] .inv$ 
   $\lambda fs P? \{ p \} \rightarrow \text{toWitness } p$ 

```

While the type signatures involved are complex, the usage is not. Finally, here is how we can automate the proof of commutativity fully:

```

 $\wedge\text{-comm} : \forall x y \rightarrow x \wedge y \equiv y \wedge x$ 
 $\wedge\text{-comm} = \forall_{\mathcal{E}}^n 2 \lambda x y \rightarrow x \wedge y \doteq y \wedge x$ 

```

With that, we now have a simple interface to a proof search library, which can be used to automate away certain tedious proofs.

Automation of simple proofs like the associativity of conjunction is all well and good, but tasks like that are more tedious than they are difficult. What about more difficult problems? In the next subsection we will look at a problem which is too complex to be solved by the simple instance-search solver we have constructed here. Instead, we will have to combine instance search with manual construction of finiteness proofs, optimisation of representation, and some other tricks. At the end of it, we will have a solver for countdown.

#### 4.4 Countdown

The Countdown problem [Hutton 2002] is a well-known puzzle in functional programming (which was apparently turned into a TV show). As a running example in this paper, we will produce a verified program which lists all solutions to a given countdown puzzle: here we will briefly explain the game and our strategy for solving it.

The idea behind countdown is simple: given a list of numbers, contestants must construct an arithmetic expression (using a small set of functions) using some or all of the numbers, to reach some target. Here's an example puzzle:

Using some or all of the numbers 1, 3, 7, 10, 25, and 50 (using each at most once), construct an expression which equals 765.

We'll allow the use of  $+$ ,  $-$ ,  $\times$ , and  $\div$ . The answer is at the bottom of this page<sup>3</sup>.

Our strategy for finding solutions to a given puzzle is to describe precisely the type of solutions to a puzzle, and then show that that type is finite. So what is a "solution" to a countdown puzzle? Broadly, it has two parts:

**A Transformation** from a list of numbers to an expression.

**A Predicate** showing that the expression is valid and evaluates to the target.

The first part is described in Figure 2.

This transformation has four steps. First (Fig. 2a) we have to pick which numbers we include in our solution. We will need to show there are finitely many ways to filter  $n$  numbers.

Secondly (Fig. 2b) we have to permute the chosen numbers. The representation for a permutation is a little trickier to envision: proving that it's finite is trickier still. We will need to rely on some of the more involved lemmas later on for this problem.

The third step (Fig. 2c) is a vector of length  $n$  of finite objects (in this case operators chosen from  $+$ ,  $\times$ ,  $-$ , and  $\div$ ). Although it is complicated slightly by the fact that the  $n$  in this  $n$ -tuple is dependent on the amount of numbers we let through in the filter in step one. (in terms of types, that means we'll need a  $\Sigma$  rather than a  $\times$ , explanations of which are forthcoming).

Finally (Fig. 2d), we have to parenthesise the expression in a certain way. This can be encapsulated by a binary tree with a certain number of leaves: proving that that is finite is tricky again.

Once we have proven that there are finitely many transformations for a list of numbers, we will then have to filter them down to those transformations which are valid, and evaluate to the target. This amounts to proving that the decidable subset of a finite set is also finite.

Finally, we will also want to optimise our solutions and solver: for this we will remove equivalent expressions, which can be accomplished with quotients. We have already introduced and described countdown: in this subsection, we will fill in the remaining parts of the solver, glue the pieces together, and show how the finiteness proofs can assist us to write the solver.




Fig. 2. The components of a transformation which makes up a Countdown candidate solution

<sup>3</sup> Answer:  $3 \times (7 \times (50 - 10) - 25)$

*Finite Vectors.* We'll start with a simple example: for both the selection (Fig. 2a) and operators (Fig. 2c) subsection, all we need to show is that a vector of some finite type is itself finite. To describe which elements to keep from an  $n$ -element list, so instance, we only need a vector of Booleans of length  $n$ . Similarly, to pick  $n$  operators requires us only to provide a vector of  $n$  operators. And we can prove in a straightforward way that a vector of finite things is itself finite.

$$\begin{aligned} \mathcal{E}!(\text{Vec}) : \mathcal{E}! A \rightarrow \mathcal{E}! (\text{Vec } A \ n) \\ \mathcal{E}!(\text{Vec}) \{n = \text{zero}\} \mathcal{E}!(A) = \mathcal{E}!(\text{Poly } \top) \\ \mathcal{E}!(\text{Vec}) \{n = \text{succ } n\} \mathcal{E}!(A) = \mathcal{E}!(A) \times \mathcal{E}!(\text{Vec}) \mathcal{E}!(A) \end{aligned}$$

We've already shown that there are finitely many booleans, the fact that there are finitely many operators is similarly simple to prove:

$$\begin{aligned} \mathcal{E}!(\text{Op}) : \mathcal{E}! \text{Op} \\ \mathcal{E}!(\text{Op}) .\text{fst} = + :: \times' :: -' :: \div' :: [] \\ \mathcal{E}!(\text{Op}) .\text{snd } + = 0, \text{ refl} \\ \mathcal{E}!(\text{Op}) .\text{snd } \times' = 1, \text{ refl} \\ \mathcal{E}!(\text{Op}) .\text{snd } -' = 2, \text{ refl} \\ \mathcal{E}!(\text{Op}) .\text{snd } \div' = 3, \text{ refl} \end{aligned}$$

*Finite Permutations.* A more complex, and interesting, step of the transformation is the first step (Fig. 2b), where we need to specify the permutation to apply to the chosen numbers.

Our first attempt at representing permutations might look something like this:

$$\begin{aligned} \text{Perm} : \mathbb{N} \rightarrow \text{Type}_0 \\ \text{Perm } n = \text{Fin } n \rightarrow \text{Fin } n \end{aligned}$$

the idea is that  $\text{Perm } n$  represents a permutation of  $n$  things, as a function from positions to positions. Unfortunately such a simple answer won't work: there are no restrictions on the operation of the function, so it could (for instance), send more than one input position into the same output.

What we actually need is not just a function between positions, but an *isomorphism* between them. In types:

$$\begin{aligned} \text{Perm} : \mathbb{N} \rightarrow \text{Type}_0 \\ \text{Perm } n = \text{Isomorphism } (\text{Fin } n) (\text{Fin } n) \end{aligned}$$

Where an isomorphism is defined as follows:

$$\begin{aligned} \text{Isomorphism} : \text{Type } a \rightarrow \text{Type } b \rightarrow \text{Type } (a \ell \sqcup b) \\ \text{Isomorphism } A B = \Sigma [ f : (A \rightarrow B) ] \Sigma [ g : (B \rightarrow A) ] (f \circ g \equiv \text{id}) \times (g \circ f \equiv \text{id}) \end{aligned}$$

While it may look complex, this term is actually composed of individual components we've already proven finite. First we have  $\text{Fin } n \rightarrow \text{Fin } n$ : functions between finite types are, as we know, finite (Theorem 3.2). We take a pair of them: pairs of finite things are *also* finite (Lemma 3.1). To get the next two components we can filter to the subobject: this requires these predicates to be decidable. We will construct a term of the following type:

$$\text{Dec } (f \circ g \equiv \text{id})$$

So can we construct such a term? Yes!

We basically need to construct decidable equality for functions between  $\text{Fin } ns$ : of course, this decidable equality is provided by the fact that such functions are themselves finite.

All in all we can now prove that the isomorphism, and by extension the permutation, is finite:

```

iso-finite :  $\mathcal{B} A \rightarrow$ 
              $\mathcal{B} B \rightarrow$ 
              $\mathcal{B} (\Sigma [f, g : (A \rightarrow B) \times (B \rightarrow A)]$ 
                $((f, g).fst \circ f, g).snd \equiv id) \times$ 
                $((f, g).snd \circ f, g).fst \equiv id))$ 
iso-finite  $\mathcal{B}\langle A \rangle \mathcal{B}\langle B \rangle =$ 
  filter
    ( $\lambda \_ \rightarrow \text{isPropEqs}$ )
    ( $\lambda \{ (f, g) \rightarrow (f \circ g) \stackrel{B}{=} id \ \&\& \ (g \circ f) \stackrel{A}{=} id \}$ )
    ( $(\mathcal{B}\langle A \rangle \mapsto \mathcal{B}\langle B \rangle) \mid \times \mid (\mathcal{B}\langle B \rangle \mapsto \mathcal{B}\langle A \rangle)$ )

```

Unfortunately this implementation is too slow to be useful. As nice and declarative as it is, fundamentally it builds a list of all possible pairs of functions between  $\text{Fin } n$  and itself (an operation which takes in the neighbourhood of  $O(n^n)$  time), and then tests each for equality (which is likely worse than  $O(n^2)$  time). We will instead use a factoriadic encoding: this is a relatively simple encoding of permutations which will reduce our time to a blazing fast  $O(n!)$ . It is expressed in Agda as follows:

```

Perm :  $\mathbb{N} \rightarrow \text{Type}_0$ 
Perm zero =  $\top$ 
Perm (suc n) =  $\text{Fin } (\text{suc } n) \times \text{Perm } n$ 

```

It is a vector of positions, each represented with a  $\text{Fin}$ . Each position can only refer to the length of the tail of the list at that point: this prevents two input positions mapping to the same output point, which was the problem with the naive encoding we had previously. And it also has a relatively simple proof of finiteness:

```

 $\mathcal{E}!\langle \text{Perm} \rangle : \mathcal{E}! (\text{Perm } n)$ 
 $\mathcal{E}!\langle \text{Perm} \rangle \{n = \text{zero}\} = \mathcal{E}!\langle \top \rangle$ 
 $\mathcal{E}!\langle \text{Perm} \rangle \{n = \text{suc } n\} = \mathcal{E}!\langle \text{Fin} \rangle \mid \times \mid \mathcal{E}!\langle \text{Perm} \rangle$ 

```

*Parenthesising.* Our next step is figuring out a way to encode the parenthesisation of the expression (Fig. 2d). At this point of the transformation, we already have our numbers picked out, we have ordered them a certain way, and we have also selected the operators we're interested in. We have, in other words, the following:

$$3 \times 7 \times 50 - 10 - 25 \quad (33)$$

Without parentheses, however, (or a religious adherence to BOMDAS) this expression is still ambiguous.

$$3 \times ((7 \times (50 - 10)) - 25) = 765 \quad (34)$$

$$(((3 \times 7) \times 50) - 10) - 25 = 1015 \quad (35)$$

The different ways to parenthesise the expression result in different outputs of evaluation.

So what data type encapsulates the “different ways to parenthesise” a given expression? That’s what we will figure out in this subsection, and what we will prove finite.

One way to approach the problem is with a binary tree. A binary tree with  $n$  leaves corresponds in a straightforward way to a parenthesisation of  $n$  numbers (Fig. 2d). This doesn't get us much closer to a finiteness proof, however: for that we will need to rely on *Dyck* words.

*Definition 10 (Dyck words).* A Dyck word is a string of balanced parentheses. In Agda, we can express it as the following:

```
data Dyck : ℕ → ℕ → Type0 where
  done : Dyck zero zero
  ⟨_ : Dyck (suc n) m → Dyck n (suc m)
  ⟩_ : Dyck n m → Dyck (suc n) m
```

A fully balanced string of  $n$  pairs of parentheses has the type `Dyck zero n`. Here are some example strings:

```
_ : Dyck 0 2          _ : Dyck 0 3
_ = ⟨ ⟩ ⟨ ⟩ done      _ = ⟨ ⟩ ⟨ ⟩ ⟨ ⟩ done
```

The first parameter on the type represents the amount of unbalanced closing parens, for instance:

```
_ : Dyck 1 2
_ = ⟩ ⟨ ⟩ ⟨ ⟩ done
```

Already Dyck words look easier to prove finite than straight binary trees, but for that proof to be useful we'll have to relate Dyck words and binary trees somehow. As it happens, Dyck words of length  $2n$  are isomorphic to binary trees with  $n - 1$  leaves, but we only need to show this relation in one direction: from Dyck to binary tree. To demonstrate the algorithm we'll use a simple tree definition:

```
data Tree : Type0 where
  leaf : Tree
  _ * _ : Tree → Tree → Tree
```

The algorithm itself is quite similar to stack-based parsing algorithms.

```
dyck→tree : Dyck zero n → Tree
dyck→tree d = go d (leaf, _)
  where
    go : Dyck n m → Vec Tree (suc n) → Tree
    go (⟨ d) ts      = go d (leaf, ts)
    go (⟩ d) (t1, t2, ts) = go d (t2 * t1, ts)
    go done (t, _)   = t
```

*Putting It All Together.* At this point we have each of the four components of the transformation defined. From this we can define what an expression is:



```

ExprTree : ℕ → Type0
ExprTree zero = ⊥
ExprTree (suc n) = Dyck 0 n × Vec Op n

Transformation : List ℕ → Type0
Transformation ns =
  Σ[ s : Subseq (length ns) ]
    let n = count s
    in Perm n × ExprTree n

```

Notice that we don't allow expressions with no numbers.

The proof that this type is finite mirrors its definition closely:

```

ℰ!(ExprTree) : ℰ! (ExprTree n)
ℰ!(ExprTree) {n = zero} = ℰ!(⊥)
ℰ!(ExprTree) {n = suc n} = ℰ!(Dyck) |×| ℰ!(Vec) ℰ!(Op)

ℰ!(Transformation) : ℰ! (Transformation ns)
ℰ!(Transformation) = ℰ!(Subseq) |Σ| λ _ → ℰ!(Perm) |×| ℰ!(ExprTree)

```

*Filtering to Correct Expressions.* We now have a way to construct, formally, every expression we can generate from a given list of numbers. This is incomplete in two ways, however. Firstly, some expressions are invalid: we should not, for instance, be able to divide two numbers which do not divide evenly. Secondly, we are only interested in those expressions which actually represent solutions: those which evaluate to the target, in other words. We can write a function which tells us if both of these things hold for a given expression like so:

<pre> eval : Tree Op ℕ → Maybe ℕ eval (leaf x) = just x eval (xs &lt; op &gt; ys) = do   x ← eval xs   y ← eval ys   x!&lt; op &gt;! y </pre>	<pre> _!&lt;_&gt;!_ : ℕ → Op → ℕ → Maybe ℕ x!&lt; + &gt;! y = just \$(x + y) x!&lt; × &gt;! y = just \$(x * y) x!&lt; - &gt;! y =   if x &lt;<sup>B</sup> y   then nothing   else just \$(x - y) x!&lt; ÷ &gt;! zero = nothing x!&lt; ÷ &gt;! suc y =   if rem x (suc y) ≡<sup>B</sup> 0   then just \$(x ÷ suc y)   else nothing </pre>
---	--

With this all together, we can finally write down the type of all solutions to a given countdown problem.

```

Solution ns n = Σ[ e : Transformation ns ] (eval (transform ns e) ≡ just n)

```

And, because the predicate here is decidable and a mere proposition, we can prove that there are finitely many solutions:

$\mathcal{E}!(\text{Solution } ns\ n)$

And we can apply this to a particular problem like so:

```
exampleSolutions :  $\mathcal{E}!(\text{Solution } [1, 3, 7, 10, 25, 50] \ 765)$ 
exampleSolutions =  $\mathcal{E}!(\langle \text{Solutions} \rangle)$ 
```

Typecheck this in Agda and it will evaluate to a list of the valid answers for that problem.

## 5 RELATED WORK

*Homotopy Type Theory.* To understand the background and subject area behind this thesis, the most important piece of work is the [Univalent Foundations Program \[2013\]](#), often simply called “the HoTT Book”.

One particular paper we relied on is [Kraus \[2015\]](#): this gave a proof (which we use) that one can eliminate out of a propositional truncation into a set provided the elimination is coherently constant.

*Agda and Cubical Type Theory.* The programming language we use in this thesis is Cubical Agda [\[Vezzosi et al. 2019\]](#). This is an implementation of Cubical Type Theory [\[Cohen et al. 2016\]](#), built as an extension to the programming language Agda [\[Norell 2008\]](#).

*Constructive Finiteness.* An excellent introduction to the topic of finiteness in a constructive setting is [Coquand and Spiwack \[2010\]](#): this paper introduced 4 notions of finiteness, called enumerated, bounded, Noetherian, and streamless. We have only looked at the first of these (enumerated, which we called “split enumerable”). The other three are interesting definitions, though: Noetherianness in particular is explored more in [Firsov et al. \[2016\]](#), and streamlessness in [Parmann \[2015\]](#).

Finite sets have long been used for proof automation in dependently typed programming languages: the most relevant example is [Firsov and Uustalu \[2015\]](#), where they are used to automate proofs in Agda. Manifest Bishop finiteness and split enumerability are the only predicates explored, however, and the setting is MLTT rather than CuTT. As a result, no proofs of equivalence are given.

Finite sets in Homotopy Type Theory in particular are explored in [Frumin et al. \[2018\]](#): that paper is quite close in subject matter to this thesis, and some of the theorems proven in this thesis are explicitly called for there. They do focus slightly more, however, on Kuratowski finite sets: these were first defined in [Kuratowski \[1920\]](#). None of the “manifest” predicates are explored in [Frumin et al. \[2018\]](#) (i.e. they focus on propositional predicates), and the only closure proofs provided are those of (non-dependent) products and sums, and surjections, on Kuratowski finite sets. As a result, there is no proof that those sets form a topos. Also, given that the work was done in HoTT and not CuTT, their univalence axiom does not compute: this means that the proof search library we have defined here would not be possible.

*Set and Topos Theory.* Our proof that finite sets form a topos follows quite closely along with the structure of section 10 of the HoTT book. This section itself is adapted from [Rijke and Spitters \[2015\]](#).

The category of finite sets in HoTT is also explored quite a bit in [Yorgey \[2014\]](#): we have formalised several of the proofs there in this thesis.

The paper [Iversen \[2018b\]](#) was essential in explaining the techniques and tricks needed to formalise category-theoretic concepts in Cubical Agda.

More generally the category of finite sets is explored in [Solov’ev \[1983\]](#), and the topic of toposes from cardinally finite sets in [Henry \[2018\]](#).

*Exhaustibility.* The twin notions of exhaustibility and omniscience were first defined in Bishop [1967]. There, they were studied as to how they applied to “constructiveness”. In functional programming, perhaps their most famous usage was in Escardó [2013]: there, Escardó shows that there are in fact sets which satisfy this principle of omniscience (or exhaustibility), *without* being finite.

*Countdown.* The countdown problem is well-known in functional programming: its first description was in Hutton [2002], but subsequent papers [Bird and Hinze 2003; Bird and Mu 2005] explored its different aspects. As far as we know, ours is the first paper to look at countdown from the dependently-typed, proof perspective.

*Generate and Test.* While the formal grounding for the API for our proof search library comes from topos theory, the less formal inspiration is from QuickCheck [Claessen and Hughes 2011]. This more general pattern of using generators in a principled way to generate data for tests is also explored in Runciman et al. [2008], which is perhaps closer in spirit to our work than QuickCheck (it’s especially close to section ??). In the setting of dependently-typed programming (and Agda in particular), O’Connor [2016] looks into the generate-and-test technique but for proofs, much like we do here. There, also, partial proof search is examined.

Finally, in order to present a usable interface to the proof search library, we used (and expanded on) some of the techniques in Allais [2019], for generic currying.

## 6 CONCLUSION

This thesis has explored finiteness in the setting of Cubical Type Theory. Hopefully, it will serve as a reasonable introduction to dependent type theory for functional programmers which is a little different from most other introductions, and which gives a taste of topics like HoTT which are on the cutting edge of the field. For those more experienced with dependent types, hopefully the thesis makes some argument as to the usefulness and importance of univalence in dependently-typed language, and demonstrates its power both theoretically and practically.

As for the theoretical contributions of the paper, we have provided a thorough accounting of many of the ways to say something is “finite” in a dependently-typed programming language, and grounded that account in topos theory. In the future we hope to see a similar exploration of the countably infinite types, and a connection of those predicates to the finite predicates.

We would also like to see more uses for finite types in dependently typed programming: termination checking seems one obvious area where they could be used more extensively.

Automated proof search libraries based on finiteness are common in the dependently-typed programming world. In this thesis, we have presented such a library which has the added power of univalence: in the future we would like to see explorations of how to improve the efficiency of the proof search, to make it more practical for larger examples. We would also like to see a similar library for the countably infinite types, which can perform partial proof search.

## REFERENCES

- Michael Abbott, Thorsten Altenkirch, and Neil Ghani. 2005. Containers: Constructing Strictly Positive Types. *Theoretical Computer Science* 342, 1 (Sept. 2005), 3–27. <https://doi.org/10.1016/j.tcs.2005.06.002>
- Guillaume Allais. 2019. Generic Level Polymorphic N-Ary Functions. In *Proceedings of the 4th ACM SIGPLAN International Workshop on Type-Driven Development - TyDe 2019*. ACM Press, Berlin, Germany, 14–26. <https://doi.org/10.1145/3331554.3342604>
- Richard Bird and Ralf Hinze. 2003. Functional Pearl Trouble Shared Is Trouble Halved. In *Proceedings of the 2003 ACM SIGPLAN Workshop on Haskell (Haskell ’03)*. ACM, New York, NY, USA, 1–6. <https://doi.org/10.1145/871895.871896>
- Richard Bird and Shin-Cheng Mu. 2005. Countdown: A Case Study in Origami Programming. *Journal of Functional Programming* 15, 05 (Aug. 2005), 679. <https://doi.org/10.1017/S0956796805005642>
- Errett Bishop. 1967. *Foundations of Constructive Analysis*. McGraw-Hill, New York.

- H. J. Boom. 1981. Further Thoughts on Abstracto. *Working Paper ELC-9, IFIP WG 2.1* (1981).
- Edwin Brady. 2013. Idris, a General-Purpose Dependently Typed Programming Language: Design and Implementation. *Journal of Functional Programming* 23, 05 (Sept. 2013), 552–593. <https://doi.org/10.1017/S095679681300018X>
- Alexander Bunkenburg. 1994. The Boom Hierarchy. In *Functional Programming, Glasgow 1993*, John T. O'Donnell and Kevin Hammond (Eds.). Springer London, 1–8. [https://doi.org/10.1007/978-1-4471-3236-3\\_1](https://doi.org/10.1007/978-1-4471-3236-3_1)
- Koen Claessen and John Hughes. 2011. QuickCheck: A Lightweight Tool for Random Testing of Haskell Programs. *SIGPLAN Not.* 46, 4 (May 2011), 53–64. <https://doi.org/10.1145/1988042.1988046>
- Cyril Cohen, Thierry Coquand, Simon Huber, and Anders Mörtberg. 2016. Cubical Type Theory: A Constructive Interpretation of the Univalence Axiom. *arXiv:1611.02108 [cs, math]* (Nov. 2016), 34. arXiv:1611.02108 [cs, math]
- Thierry Coquand and Arnaud Spiwack. 2010. Constructively Finite?. In *Contribuciones Científicas En Honor de Mirian Andrés Gómez*. Universidad de La Rioja, 217–230.
- Nils Anders Danielsson. 2012. Bag Equivalence via a Proof-Relevant Membership Relation. In *Interactive Theorem Proving (Lecture Notes in Computer Science)*. Springer, Berlin, Heidelberg, 149–165. [https://doi.org/10.1007/978-3-642-32347-8\\_11](https://doi.org/10.1007/978-3-642-32347-8_11)
- Martin Escardo. 2007. Infinite Sets That Admit Fast Exhaustive Search. In *22nd Annual IEEE Symposium on Logic in Computer Science (LICS 2007)*. IEEE, Wrocław, Poland, 443–452. <https://doi.org/10.1109/LICS.2007.25>
- Martín H. Escardó. 2013. Infinite Sets That Satisfy the Principle of Omniscience in Any Variety of Constructive Mathematics. *The Journal of Symbolic Logic* 78, 3 (Sept. 2013), 764–784. <https://doi.org/10.2178/jsl.7803040>
- Denis Firsov and Tarmo Uustalu. 2015. Dependently Typed Programming with Finite Sets. In *Proceedings of the 11th ACM SIGPLAN Workshop on Generic Programming - WGP 2015*. ACM Press, Vancouver, BC, Canada, 33–44. <https://doi.org/10.1145/2808098.2808102>
- Denis Firsov, Tarmo Uustalu, and Niccolò Veltri. 2016. Variations on Noetherianness. *Electronic Proceedings in Theoretical Computer Science* 207 (April 2016), 76–88. <https://doi.org/10.4204/EPTCS.207.4> arXiv:1604.01186
- Dan Frumin, Herman Geuvers, Léon Gondelman, and Niels van der Weide. 2018. Finite Sets in Homotopy Type Theory. In *Proceedings of the 7th ACM SIGPLAN International Conference on Certified Programs and Proofs (CPP 2018)*. ACM, New York, NY, USA, 201–214. <https://doi.org/10.1145/3167085>
- Michael Hedberg. 1998. A Coherence Theorem for Martin-Löf's Type Theory. *Journal of Functional Programming* 8, 4 (July 1998), 413–436. <https://doi.org/10.1017/S0956796898003153>
- Simon Henry. 2018. On Toposes Generated by Cardinal Finite Objects. *Mathematical Proceedings of the Cambridge Philosophical Society* 165, 2 (Sept. 2018), 209–223. <https://doi.org/10.1017/S0305004117000408> arXiv:1505.04987
- Jason Z. S. Hu and Jacques Carette. 2020. Proof-Relevant Category Theory in Agda. *arXiv:2005.07059 [cs]* (May 2020). arXiv:2005.07059 [cs]
- Graham Hutton. 2002. The Countdown Problem. *J. Funct. Program.* 12, 6 (Nov. 2002), 609–616. <https://doi.org/10.1017/S0956796801004300>
- Frederik Hanghøj Iversen. 2018a. Fredefox/Cat.
- Frederik Hanghøj Iversen. 2018b. *Univalent Categories: A Formalization of Category Theory in Cubical Agda*. Master's Thesis. Chalmers University of Technology, Göteborg, Sweden.
- Nicolai Kraus. 2015. The General Universal Property of the Propositional Truncation. *arXiv:1411.2682 [math]* (Sept. 2015), 35 pages. <https://doi.org/10.4230/LIPIcs.TYPES.2014.111> arXiv:1411.2682 [math]
- Casimir Kuratowski. 1920. Sur la notion d'ensemble fini. *Fundamenta Mathematicae* 1, 1 (1920), 129–131.
- Ulf Norell. 2008. Dependently Typed Programming in Agda. In *Proceedings of the 6th International Conference on Advanced Functional Programming (AFP'08)*. Springer-Verlag, Heijten, The Netherlands, 230–266.
- Liam O'Connor. 2016. Applications of Applicative Proof Search. In *Proceedings of the 1st International Workshop on Type-Driven Development (TyDe 2016)*. ACM, New York, NY, USA, 43–55. <https://doi.org/10.1145/2976022.2976030>
- Erik Parmann. 2015. Investigating Streamless Sets. In *20th International Conference on Types for Proofs and Programs (TYPES 2014) (Leibniz International Proceedings in Informatics (LIPIcs), Vol. 39)*, Hugo Herbelin, Pierre Letouzey, and Matthieu Sozeau (Eds.). Schloss Dagstuhl–Leibniz-Zentrum fuer Informatik, Dagstuhl, Germany, 187–201. <https://doi.org/10.4230/LIPIcs.TYPES.2014.187>
- Egbert Rijke and Bas Spitters. 2015. Sets in Homotopy Type Theory. *Mathematical Structures in Computer Science* 25, 5 (June 2015), 1172–1202. <https://doi.org/10.1017/S09560129514000553>
- Colin Runciman, Matthew Naylor, and Fredrik Lindblad. 2008. SmallCheck and Lazy SmallCheck: Automatic Exhaustive Testing for Small Values. In *In Haskell'08: Proceedings of the First ACM SIGPLAN Symposium on Haskell*, Vol. 44. ACM, 37–48.
- S. V. Solov'ev. 1983. The Category of Finite Sets and Cartesian Closed Categories. *Journal of Soviet Mathematics* 22, 3 (June 1983), 1387–1400. <https://doi.org/10.1007/BF01084396>
- The Coq Development Team. 2020. The Coq Proof Assistant, Version 8.11.0. Zenodo. <https://doi.org/10.5281/zenodo.3744225>
- The Univalent Foundations Program. 2013. *Homotopy Type Theory: Univalent Foundations of Mathematics*. <https://homotopytypetheory.org/book>, Institute for Advanced Study.

- Andrea Vezzosi, Anders Mörtberg, and Andreas Abel. 2019. Cubical Agda: A Dependently Typed Programming Language with Univalence and Higher Inductive Types. *Proc. ACM Program. Lang.* 3, ICFP (July 2019), 87:1–87:29. <https://doi.org/10.1145/3341691>
- Philip Wadler. 2015. Propositions As Types. *Commun. ACM* 58, 12 (Nov. 2015), 75–84. <https://doi.org/10.1145/2699407>
- Brent Abraham Yorgey. 2014. *Combinatorial Species and Labelled Structures*. Ph.D. Dissertation. University of Pennsylvania, Pennsylvania.